


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L. 1.000

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice



PAPER J.OFT

18

sped. in abb. post. gr. 1170

Anno I - N° 18 - 26 ottobre 1984



**Listatore
Dardi**



**Filetto
Pixelator**



**Numeri in scala
Cubi spaziali**



**Istogrammi a barre
SuperSprite**



**Space Patrol
VIC Organetto
Bee Trap**

Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numero seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	█
{G2}	2	█
{G3}	3	█
{G4}	4	█
{G5}	5	█
{G6}	6	█
{G7}	7	█
{G8}	8	█
{SG1}	CAPS SHIFT 1	█
{SG2}	CAPS SHIFT 2	█
{SG3}	CAPS SHIFT 3	█
{SG4}	CAPS SHIFT 4	█
{SG5}	CAPS SHIFT 5	█
{SG6}	CAPS SHIFT 6	█
{SG7}	CAPS SHIFT 7	█
{SG8}	CAPS SHIFT 8	█

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
A	A	Simbolo grafico definito nel programma in uso.
B	B	
C	C	
D	D	
E	E	
F	F	
G	G	
H	H	
I	I	
J	J	
K	K	
L	L	
M	M	
N	N	
O	O	
P	P	
Q	Q	
R	R	
S	S	
T	T	
U	U	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [< >] deve essere

premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL. Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME	█	{CYN}	CTRL 4	█	[<7>]	G 7	█
{HOME}	CLR/HOME	█	{PUR}	CTRL 5	█	[<8>]	G 8	█
{SU}	SHIFT CRSR	█	{GRN}	CTRL 6	█	{F1}	f1	█
{GIU'}	CRSR	█	{BLU}	CTRL 7	█	{F2}	f2	█
{SIN}	SHIFT ← CRSR →	█	{YEL}	CTRL 8	█	{F3}	f3	█
{DES}	← CRSR →	█	[<1>]	G 1	█	{F4}	f4	█
{RVS}	CTRL 9	█	[<2>]	G 2	█	{F5}	f5	█
{OFF}	CTRL 0	█	[<3>]	G 3	█	{F6}	f6	█
{BLK}	CTRL 1	█	[<4>]	G 4	█	{F7}	f7	█
{WHT}	CTRL 2	█	[<5>]	G 5	█	{F8}	f8	█
{RED}	CTRL 3	█	[<6>]	G 6	█			█



		4	Listatore di L. Anderson trad. e adatt. di M. Cerofolini
paddle		6	Dardi di M. Daoust trad. e adatt. di M. Cerofolini
extended basic	 TI-99/4A	8	Filetto di D. Massè trad. e adatt. di E. Re Garbagnati
basic	 TI-99/4A	11	Pixelator di M. Bianchi A. Carini e D. Della Rossa
16/48K		14	Numeri in scala di M. Savin trad. e adatt. di C. Panzalis
16/48K		16	Cubi spaziali di D. Parkinson trad. e adatt. di P. Maffei
		18	Istogrammi a barre di R. A. Hicks trad. e adatt. di U. Barzaghi
		19	SuperSprite di N. Sullivan trad. e adatt. di M. Sciancalepore
joystick		22	Space Patrol di S. Alcàntara trad. e adatt. di M. Anticoli e S. Albarelli
		25	Vic Organetto di M. Anticoli e S. Albarelli
joystick		28	Bee Trap di R. L. Lykins trad. e adatt. di S. Alberelli e M. Anticoli

J. soft s.r.l.
**DIREZIONE, REDAZIONE,
 AMMINISTRAZIONE**
 Via Rosellini, 12
 20124 MILANO
 Tel. (02) 68.88.228

DIRETTORE RESPONSABILE:
 Pietro Dell'Orco

COORDINAMENTO TECNICO:
 Riccardo Paolillo

REDAZIONE:
 Lucio Bragagnolo
 Mauro Cristuib. Grizzi

GRAFICA E IMPAGINAZIONE:

Margherita La Noce
 Ivana Rossi
 Raffaella Toffolatti

FOTOCOMPOSIZIONE:

d&b Via Vignola, 5
 Tel. 02/59.85.08
 20133 MILANO

CONTABILITÀ:

Giulia Pedrazzini
 Flavia Bonaiti

**AUTORIZZAZIONE ALLA
 PUBBLICAZIONE:**

Tribunale di Milano n° 200
 del 14.04.1984

STAMPA:

Elcograf
 Beverate (CO)

PUBBLICITÀ

Concessionario per l'Italia e l'Estero Reina s.r.l.
 Via Washington, 50
 20046 MILANO
 Tel. (02) 49.88.066 (5 linee R.A.)
 Tlx. 316213 REINA I

Concessionario esclusivo per la
 DIFFUSIONE in Italia e Estero:
 SODIP - Via Zuretti, 25
 20125 MILANO

Spedizione in abbonamento
 postale Gruppo II/70
 Prezzo della rivista L. 1.000
 Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI
 RIPRODUZIONE O TRADUZIONE
 DEGLI ARTICOLI PUBBLICATI
 SONO RISERVATI



Listatore

Potete ancora capire un programma che avete scritto sei mesi fa?

Vi ricordate il trucco che avete in quella lunga linea BASIC che c'è nel vostro programma? Questo programma trasforma il listing del vostro programma Applesoft in un formato strutturato e di facile lettura in particolare:

- separa nettamente i commenti dalle istruzioni.
- omette i due punti e gli statement LET.
- indenta gli statement FOR NEXT globalmente.
- spezza logicamente le linee molto lunghe.

Copiate accuratamente i due programmi:

1) LISTATORE

```

1  HOME : INPUT "SLOT STAMPANTE?
   ";SL: GOTO 82
2  P = P + 1:B = PEEK (P): RETURN
4  D = 0: GOSUB 6: PRINT S$: RETURN
6  LC = LC + 1: IF LC < LP THEN
   RETURN
7  GOSUB 111:LC = 6:PC = PC + 1:
   PRINT S$: PRINT BB$;LB$;"(CONTI
   NUA)"
9  FOR K = LP TO 63: PRINT S$: N
   EXT
11 H$(4) = "PAG. " + STR$(PC):
   FOR K = 1 TO 4:E = INT ((LL -
   LEN (H$(K))) / 2) + 1: PRINT M$
   ; LEFT$(BB$,E);H$(K): NEXT :K =
   FRE (0): PRINT S$: IF NOT D T
   HEN RETURN
13 N$ = STR$( VAL (N$)):K = L
   EN (N$)
14 C$ = RIGHT$( ( LEFT$(LB$, (6
   - K)) + CHR$(91) + N$ + CHR$(
   93) + S$),8) + RIGHT$(C$, (L
   EN (C$) - 8)):K = FRE (0): RETU
   RN
16 GOSUB 2:D = B: GOSUB 2:D = 2
   56 * B + D: IF D > 0 THEN GOTO
   27
18 GOSUB 111: GOSUB 4: GOSUB 4:
   GOSUB 6: PRINT M$;LB$;"FINE LIS
   TING"

```

2) CREA LIST

Dopo aver controllato l'esattezza della digitazione, eseguite il programma CREA LIST. Questo programma crea un file in formato TEXT dal nome LIST. A questo punto avete sul dischetto sia il file in formato text dal nome LIST che il programma LISTATORE. Non vi servirà più il programma CREA LIST che potrete memorizzare su un altro dischetto.

Il programma da listare va caricato per primo. Quindi si richiama il file LIST col comando EXEC LIST. Questo comando carica e mette in esecuzione il programma LISTATORE che effettua effettivamente la stampa del primo programma caricato in memoria.

```

20 GOSUB 4: GOSUB 4: GOSUB 6: P
   RINT M$;"LUNGHEZZA PROGRAMMA = "
   ;( PEEK (211) - PEEK (209)) * 2
   56 + PEEK (210) - PEEK (208);"
   BYTES, TOTALE DI ";TN;" NUMERI
   DI LINEA"
21 GOSUB 4: GOSUB 6: PRINT M$;(
   TS - TR);" TOTALE STATEMENTS NON
   -REM, ";TR;" TOTALE REMARKS": GO
   SUB 4: GOSUB 6: PRINT M$;"END"
23 PRINT CHR$(4);"PR# 0": POK
   E 33,40: HOME : VTAB 12: HTAB 11
   : INVERSE : PRINT " FINE LISTING
   ": NORMAL
25 POKE 105, PEEK (210): POKE 1
   06, PEEK (211): POKE 107, PEEK (
   210): POKE 108, PEEK (211): POKE
   109, PEEK (210): POKE 110, PEEK
   (211): POKE 111, PEEK (115): PO
   KE 112, PEEK (116): POKE 103, PE
   EK (208): POKE 104, PEEK (209):
   POKE 175, PEEK (210): POKE 176,
   PEEK (211): END
27 TN = TN + 1: GOSUB 2:D = B: G
   OSUB 2:K = B * 256 + D:D = LEN
   ( STR$(K)):N$ = RIGHT$( ( LEFT
   $(LB$, (7 - D)) + STR$(K) + "
   "),8)
29 TS = TS + 1:D = 0: GOSUB 2: I
   F B = 58 THEN B = 178
30 IF B = 178 AND NOT RF THEN

```

```

GOSUB 4: GOTO 34
31 IF B = 178 AND RF THEN GOTO
34
32 IF RF THEN RF = 0: GOSUB 4
34 IF B = 0 GOTO 54
35 IF B > 127 THEN B = B - 127:
GOTO 41
36 IF B = 34 THEN QF = - QF
37 IF B = 58 AND NOT RF AND QF
< 1 THEN TS = TS + 1:G$ = G$ +
" :": GOTO 54
38 IF B < 32 THEN B = B + 64:G$
= G$ + CHR$(124) + CHR$(B):
B = 124
39 G$ = G$ + CHR$(B): GOSUB 2:
GOTO 34
41 IF B > 107 THEN G$ = G$ + "
?! ": GOSUB 2: GOTO 34
43 IF B = 51 THEN TR = TR + 1:R
F = 1:RS = 1
44 IF B = 2 THEN FF = 1
45 IF B = 69 THEN CF = 1:G$ = G
$ + T$(B): GOTO 54
46 IF B = 4 THEN DF = 1
47 IF B = 43 THEN GOSUB 2: GOT
O 34
48 IF B = 3 THEN FS = FS - 1: I
F FS < 0 THEN FS = 0
49 G$ = G$ + T$(B): GOSUB 2: GOT
O 34
51 SF = 0:RS = RS + 1: IF RS > 2
THEN RS = 2:
52 IF DFANSRS > 1 THEN RS = 1
54 K = IM * (FS + CS + RS):E = K
+ 13: IF K > 0 THEN G$ = LEFT$(
(BB$,K) + G$
56 IF NOT D THEN C$ = N$ + G$
57 IF D THEN C$ = LB$ + G$
59 K = LEN (C$) - LL: IF K < 1
GOTO 73
60 G$ = RIGHT$(C$,K):C$ = LEF
T$(C$,LL):SF = 1: IF DF GOTO 65

62 D = LL
63 IF MID$(C$,D,1) = S$ GOTO
71
64 D = D - 1: IF D > E THEN GOT
O 63
65 D = LL
66 K = ASC ( MID$(C$,D,1)): IF
K < 42 OR K > 47 THEN GOTO 69
67 IF DF AND K = 44 THEN GOTO
71
68 IF NOT DF AND K < > 46 GOT
O 71
69 D = D - 1: IF D > E GOTO 66
70 GOTO 73
71 K = LL - D: IF K > 0 THEN G$

```

```

= RIGHT$(C$,K) + G$:C$ = LEFT
$(C$,D)
73 GOSUB 6:K = LEN (C$): IF SF
= 0 OR K < 2 OR RF THEN 76
74 IF MID$(C$,K,1) = S$ THEN
C$ = LEFT$(C$, (K - 1)) + CHR$(
(95)
75 IF LEN (G$) > 2 AND LEFT$(
(G$,1) = S$ THEN G$ = CHR$(95)
+ RIGHT$(G$, LEN (G$) - 1)
76 GOSUB 111:K = LEN (C$): PRI
NT M$; LEFT$(C$,8);: GOSUB 112:
PRINT RIGHT$(C$, (K - 8)):K =
FRE (0): IF SF THEN D = 1: GOTO
51
77 QF = - 1:RS = 0:DF = 0: IF F
F THEN FS = FS + 1:FF = 0
78 D = 0: IF CF THEN CS = CS + 1
:CF = 0
79 SF = 0:G$ = "": IF B = 0 THEN
CS = 0: GOTO 16
80 GOSUB 2:D = 1: GOTO 34
82 DIM T$(107),H$(4)
85 P = 2048:B = 0:RS = 0:CS = 0:
FS = 0:RF = 0:CF = 0:FF = 0:DF =
0:SF = 0:QF = - 1:LL = 80:LP =
60:IM = 4:E = 0:TN = 0:TS = 0:T
R = 0:S$ = " "
86 C$ = "":N$ = "":G$ = "":M$ =
"":H$(0) = "":LB$ = " " :B
B$ = "

"
87 DATA "End","For ","Next ",
"Data ","Input ","Del ","Dim ","
Read ","Gr ","Text ","Pr# ","In#
","Call ","Plot ","Hlin ","Vlin
","Hgr2","Hgr","Hcolor = ","Hpl
ot ","Draw ","Xdraw ","Htab ","H
ome ","Rot = "
88 DATA "Scale = ","Shload"
,"Trace","Notrace","Normal","Inv
erse","Flash","Color = ","Pop","
Vtab","Himem : ","Lomem : ","One
rr ","Resume ","Recall ","Store
","Speed = "," ","Goto ","Run","
If ","Restore","& ","Gosub ","Re
turn"
89 DATA "REM ----- "
90 DATA "Stop","On ","Wait ",
"Load ","Save ","Def ","Poke ","
Print ","Cont","List ","Clear","
Get ","New","Tab("," To ","Fn ",
"SpC("," Then "," At ","Not ","S
tep ","+","-","*"
91 DATA "/" ,"^"," And "," Or "
,">"," = ","<","Sgn","Int","Abs"

```

```

,"Usr","Fre","Scrn(","Pdl","Pos"
,"Sqr","Rnd","Log","Exp","Cos","
Sin","Tan","Arctan","Peek","Len"
,"Str$","Val","Asc","Chr$","Left
$","Rigjt$","Mid$"
92 FOR K = 1 TO 107: READ T$(K)
: NEXT
94 HOME : VTAB 3: FLASH : PRINT
" RUN 23 ";: NORMAL : PRINT "RE
SETTA I PUNTATORI DOPO RESET"
95 PRINT : PRINT "INIZIO DEL PR
OGRAMMA: "; PEEK (209) * 256 +
PEEK (208): PRINT " FINE DEL PR
OGRAMMA: "; PEEK (211) * 256 +
PEEK (210): PRINT " FINE D
I 'LISTATORE': "; PEEK (176) * 2
56 + PEEK (175)
97 PRINT : INPUT "NOME PROGRAMM
A: ";H$(1): INPUT " PROGRAMMATO
RE: ";H$(2): INPUT " DATA
: ";H$(3): PRINT
98 PRINT : PRINT "LUNGHEZZA PAG
. 60 LINEE,VUOI CAMBIARE?": GET
H$(0): IF H$(0) = "S" THEN INPU
T " LUNGH. PAG.: ";LP: IF LP >
62 THEN GOTO 98
99 PRINT : PRINT "MARGINE SINIS
TRO = 0. LO VUOI?": GET H$(0): I
F H$(0) = "S" THEN INPUT " MAR
GINE : ";K: IF K > 0 AND K < 49
THEN M$ = LEFT$(BB$,K):LL = LL
- K
100 PRINT : PRINT "RIENTRO SPAZ
I = 4, VUOI CAMBIARE?": GET H$(
0): IF H$(0) = "S" THEN INPUT "
SPAZI: ";IM: IF IM < 0 OR IM >
12 THEN GOTO 100
102 HOME : INVERSE : PRINT " CA
RTA AD INIZIO FOGLIO ": PRINT "

```

```

POI " : PRI
NT " ACCENDI LA STAMPANTE"
: NORMAL : PRINT : GET H$(0)
104 HOME : PRINT CHR$(4);"PR#
";SL: POKE 33,30
106 PRINT CHR$(9)"66N" CHR$(
9)"I"
109 LC = 6:PC = 1: GOSUB 11: GOT
O 16
111 RETURN
112 RETURN
210 D$ = CHR$(4):C$ = CHR$(1
3) + D$
215 PRINT C$;"OPEN LIST"
220 PRINT C$;"WRITE LIST"
221 PRINT "POKE 208,PEEK(103)"
222 PRINT "POKE 209,PEEK(104)"
223 PRINT "POKE 210,PEEK(175)"
224 PRINT "POKE 211,PEEK(176)"
230 PRINT "POKE 104,PEEK(211)"
235 PRINT "IF PEEK(210)<254 THE
N POKE 103,(PEEK(210)+1)"
240 PRINT "IF PEEK(210)>253 THE
N POKE 103,(PEEK(210)-254):POKE
104,(PEEK(211)+1)
245 PRINT "POKE (PEEK(210)+PEEK
(211)*256-2),0"
250 PRINT "POKE (PEEK(210)+PEEK
(211)*256-1),0"
255 PRINT "POKE (PEEK(210)+PEEK
(211)*256),0"
260 PRINT "POKE (PEEK(210)+PEEK
(211)*256+1),0"
265 PRINT "POKE (PEEK(210)+PEEK
(211)*256+2),0"
270 PRINT "RUN LISTATORE"
280 PRINT D$;"CLOSE"

```

Dardi

Con questo gioco vengono messi alla prova la rapidità di riflessi e la precisione nella mira di due concorrenti, i quali potranno comodamente divertirsi in casa senza rischiare di coprire di fori le pareti o di trafiggere qualche incauto amico.



paddle

Il bersaglio comparirà nello schermo in alto, mentre i dardi che devono essere lanciati passeranno in basso. Ad ogni turno ciascun giocatore ha a disposizione tre dardi, che lancerà premendo il bottone della paddle. Il punteggio per ogni lancio varia da 10 a 50 a se-

conda del settore del bersaglio colpito dal dardo. Vince il concorrente che totalizza per primo 210 punti esatti.

Attenzione: qualora con gli ultimi lanci si superasse questo limite il punteggio non aumenterà.

```

34 GOSUB 30000
35 P = 230:D = 5
40 CALL - 936
41 HTAB 15: PRINT "FRECCE": PRINT : PRINT
56 INPUT "DAI IL NOME DEL PRIMO GIOCATORE ";N1$
58 INPUT "DAI IL NOME DEL SECONDO GIOCATORE ";N2$
70 REM MAIN
80 GOSUB 1000
100 S1 = 0:S2 = 0:N = 1
105 CALL - 936
106 GOSUB 4500
130 FOR X = 0 TO 39
132 COLOR= 15: PLOT X,39
135 IF PEEK ( - 16287 + TURN) > 127 THEN GOSUB 2000
148 COLOR= 0: PLOT X,39
155 NEXT X
200 GOTO 130
900 END
1000 REM ROUTINE PER DISEGNARE IL BERSAGLIO
1010 GR
1020 COLOR= 9
1030 FOR F = 10 TO 13: VLIN 0,3 AT F: VLIN 0,3 AT F + 15: NEXT F
1035 COLOR= 6: FOR F = 14 TO 16: VLIN 0,3 AT F: VLIN 0,3 AT F + 8: NEXT F
1040 COLOR= 13: FOR F = 17 TO 18: VLIN 0,3 AT F: VLIN 0,3 AT F + 3: NEXT F
1045 COLOR= 1: VLIN 0,3 AT 19
1050 COLOR= 8: HLIN 0,39 AT 0
1095 RETURN
2000 FOR Y = 39 TO 0 STEP - 1
2003 COLOR= 15
2005 PLOT X,Y
2030 IF SCRN( X,Y - 1) < > 0 THEN 2050
2038 COLOR= 0: PLOT X,Y
2040 NEXT Y
2050 N = N + 1
2055 GOSUB 7000
2060 IF TURN = 1 THEN 3000
2065 IF TURN = 0 THEN 4000
2490 X = 0
2500 RETURN

```

```

3000 REM TURNO DEL GIOCAT. 2
3020 IF N = 4 THEN TURN = 0: GOSUB 7050
3022 IF N = 4 THEN GOSUB 1000
3025 IF N = 4 THEN N = 1
3030 GOSUB 5000
3050 S2 = S2 + HIT
3052 IF S2 > 210 THEN S2 = S2 - HIT
3055 GOSUB 4500
3400 GOSUB 2490
4000 REM TURNO DEL GIOCAT. 1
4020 IF N = 4 THEN TURN = 1: GOSUB 7050
4022 IF N = 4 THEN GOSUB 1000
4025 IF N = 4 THEN N = 1
4030 GOSUB 5000
4050 S1 = S1 + HIT
4052 IF S1 > 210 THEN S1 = S1 - HIT
4055 GOSUB 4500
4400 GOTO 2490
4490 REM STAMPA IL PUNTEGGIO
4500 CALL - 936
4510 VTAB 21: PRINT N1$;" = ";S1;: HTAB 20: PRINT N2$;" = ";S2
4515 IF S1 = 210 OR S2 = 210 THEN GOSUB 6000
4518 VTAB 23: HTAB 10
4520 IF TURN = 0 THEN PRINT "T OCCA A ";N1$;" TIRI ";N;" "
4522 IF TURN = 1 THEN PRINT "T OCCA A ";N2$;" TIRI ";N;" "
4550 RETURN
5000 REM CALCOLA IL PUNTEGGIO
5010 HIT = 0
5015 IY = 1
5020 IF Y = 5 THEN IY = 2
5022 IF Y = 6 THEN IY = 3
5023 IF Y = 7 THEN IY = 4
5024 IF Y = 8 THEN IY = 5
5030 IF SCRN( X,Y - IY) = 8 THEN HIT = 0
5032 IF SCRN( X,Y - IY) = 9 THEN HIT = 10
5034 IF SCRN( X,Y - IY) = 6 THEN HIT = 20
5036 IF SCRN( X,Y - IY) = 13 THEN HIT = 30

```

```

5038 IF SCRN( X,Y - IY) = 1 TH
EN HIT = 50
5100 RETURN
5990 REM FINE GIOCO
6000 FOR S = 1 TO 50: GOSUB 700
0: NEXT S
6003 VTAB 23: INPUT "VUOI GIOCA
RE ANCORA? ";A$
6005 IF S2 = 210 THEN TURN = 0
6007 IF S1 = 210 THEN TURN = 1
6010 IF LEFT$(A$,1) = "S" THE
N 80
6020 TEXT : CALL - 936: END

```

```

6990 REM SUONO
7000 POKE PITCH,P: POKE TIME,D:
CALL 768: RETURN
7050 FOR KK = 1 TO 4: GOSUB 700
0: NEXT KK
7100 RETURN
30000 POK$ = "173,048,192,136,20
8,004,198,001,240,008,202,208,24
6,166,000,076,000,003,096"
30010 FOR PK = 0 TO 18: POKE 76
8 + PK, VAL ( MID$( POK$,PK * 4
+ 1,3)): NEXT PK
30020 PITCH = 0:TIME = 1: RETURN

```



Filetto

extended basic

Persino alla Tv in un noto quiz è stata riproposta l'ennesima versione di questo divertentissimo gioco di intelligenza, ed allora perché noi dovremmo essere da meno?

Eccovi allora un'ottima versione per il vostro TI-99 in linguaggio EXTENDED-BASIC che vi permetterà di misurarvi con un computer più agguerrito che mai, alla ricerca dell'allineamento di quattro qualsiasi pedine personali.

Inutile dare consigli strategici per questo divertentissimo gioco, ma opportuno segnalarvi al-

meno quali tasti utilizzare per effettuare le vostre mosse.

Il computer vi richiederà, dopo aver "sorteggiato" il giocatore avente diritto alla prima mossa, di volta in volta la colonna in cui desiderate depositare la vostra pedina e voi non avrete altro da fare che inserirne il numero e convalidare la mossa premendo il tasto ENTER. Dopodiché penserà a tutto il computer, anche ad indicare la realizzazione di un "Filetto", e persino a battervi, garantito!

```

148 CALL CLEAR
150 DIM A(5,7),N(20)
160 ON WARNING NEXT
170 N(1)=.01
180 N(2)=.1
190 N(3)=10
200 N(4)=1000
210 N(6)=1
220 N(11)=100
230 N(16)=10000
240 CALL CHAR(104,"01010303070F3
EFC8080C0C0E0F07C3FFC3E0F0703030
1013F7CF0E0C0C08080")
250 CALL COLOR(10,13,4)
260 CALL CHAR(112,"804020181C0E0
703010204183870E0C003070E1C18204
080C0E0703818040201")
270 CALL COLOR(11,5,12)

```

```

280 ON ERROR 1880
290 JC=14 :: SC=2 :: K=1
300 GOSUB 2270
310 DISPLAY AT(1,10)ERASE ALL:"#
FILETTO #" :: GOSUB 320 :: GOTO
410
320 CALL SOUND(100,494,2)
330 CALL SOUND(100,440,2)
340 CALL SOUND(400,494,2)
350 CALL SOUND(100,440,2)
360 CALL SOUND(100,392,2)
370 CALL SOUND(100,349,2)
380 CALL SOUND(200,330,2)
390 CALL SOUND(400,294,2)
400 RETURN
410 DISPLAY AT(2,10):"=====
="
420 DISPLAY AT(10,5):"SCEGLI IL

```



```

LIVELLO DI GIOCO:"
430 DISPLAY AT(13,5):"0 PRINC
IPIANTE": " 1 ESPERTO"
440 ACCEPT AT(17,15):ND
450 IF ND<>0 AND ND<>1 THEN 440

460 REM GIOCO
470 GOSUB 1910
480 DISPLAY AT(3,0):"OBIETTIVO:
ALLINEARE 4 ":"GETTONI, IN QUALSI
ASI VERSO."
490 DISPLAY AT(6,0):"SORTEGGIO P
ER CHI INIZIA..."

500 FOR I=1 TO 1000 :: NEXT I
510 RANDOMIZE
520 JO=INT(RND*2)+1
530 IF JO=1 THEN 720
540 REM MOSSA GIOCATORE
550 JC=13 :: SC=4 :: GOSUB 2270

560 C=C+1
570 IF C=36 THEN 1510
580 DISPLAY AT(3,0):"TOCCA A TE"
,"NUMERO DI COLONNA, PREGO"
590 DISPLAY AT(6,0):" "
600 ACCEPT AT(4,26)BEEP:X
610 IF X<1 OR X>7 THEN 600
620 IF G(X)<5 THEN 650
630 DISPLAY AT(6,0):"COLONNA COM
PLETA:RITENTA!"
640 FOR I=1 TO 200 :: NEXT I ::
DISPLAY AT(6,0):" " :: GOTO 600

650 Y=5-G(X)
660 CAR=104
670 G(X)=G(X)+1
680 A(Y,X)=1
690 GOSUB 1450
700 IF L(X)=1 THEN 1570
710 GOTO 720
720 REM MOSSA COMPUTER
730 C=C+1
740 B=0
750 M=-10000
760 IF C=36 THEN 1510
770 GOSUB 1820
780 FOR X=1 TO 7
790 TT=0 :: GOSUB 860
800 IF ND THEN GOSUB 1400
810 IF T<M THEN 840
820 M=T
830 XM=X
840 NEXT X
850 GOTO 1290
860 L(X)=0
870 T=0
880 Y=5-G(X)
890 IF (Y=0 AND TT=0)OR(Y<0 AND

```

```

TT=1) THEN T=-100000 :: RETURN
900 IF Y=5 THEN 950
910 S=0
920 FOR J=1 TO 4 :: S=S+A(J,X)::
NEXT J
930 GOSUB 1740
940 IF Y=1 THEN 970
950 S=0
960 FOR J=2 TO 5 :: S=S+A(J,X)::
NEXT J
970 GOSUB 1740
980 IF X>4 THEN 1050
990 FOR D=1 TO X
1000 S=0
1010 FOR I=D TO D+3 :: S=S+A(Y,I
):: NEXT I
1020 GOSUB 1740
1030 NEXT D
1040 GOTO 1100
1050 FOR D=X-3 TO 4
1060 S=0
1070 FOR I=D TO D+3 :: S=S+A(Y,I
):: NEXT I
1080 GOSUB 1740
1090 NEXT D
1100 AA=X+Y
1110 IF Y=5 OR AA>8 OR AA<5 THEN
1150
1120 S=0
1130 FOR J=1 TO 4 :: I=AA-J :: S
=S+A(J,I):: NEXT J
1140 GOSUB 1740
1150 IF Y=1 OR AA>9 OR AA<6 THEN
1190
1160 S=0
1170 FOR J=2 TO 5 :: I=AA-J :: S
=S+A(J,I):: NEXT J
1180 GOSUB 1740
1190 AA=X-Y
1200 IF Y=5 OR AA<0 OR AA>3 THEN
1240
1210 S=0
1220 FOR J=1 TO 4 :: I=AA+J :: S
=S+A(J,I):: NEXT J
1230 GOSUB 1740
1240 IF Y=1 OR AA<-1 OR AA>2 THE
N 1280
1250 S=0
1260 FOR J=2 TO 5 :: I=AA+J :: S
=S+A(J,I):: NEXT J
1270 GOSUB 1740
1280 RETURN
1290 X=XM
1300 DISPLAY AT(4,12):X
1310 FOR I=1 TO 200 :: NEXT I
1320 CAR=112
1330 Y=5-G(X)
1340 G(X)=G(X)+1

```

```

1350 A(Y,X)=5
1360 GOSUB 1450
1370 B=1 :: TT=0
1380 GOSUB 860
1390 GOTO 540
1400 REM VALUTAZIONE
      CASI PRECEDENTI
1410 Y=Y-1 :: TT=1
1420 GOSUB 890
1430 Y=Y+1
1440 RETURN
1450 REM STAMPA GETTONI
1460 CALL HCHAR(7+3*Y,3*X,CAR)
1470 CALL HCHAR(7+3*Y,1+3*X,CAR+
1)
1480 CALL HCHAR(8+3*Y,3*X,CAR+2)
1490 CALL HCHAR(8+3*Y,1+3*X,CAR+
3)
1500 RETURN
1510 REM PAREGGIO
1520 JC=14 :: SC=2 :: GOSUB 2270
1530 DISPLAY AT(3,0):"PAREGGIO.
NON MALE!!", " ":" ":" " :: GOSU
B 320
1540 FOR I=1 TO 1000 :: NEXT I
1550 GOSUB 1910
1560 GOTO 490
1570 REM VITTORIA GIOCATORE
1580 JC=14 :: SC=2 :: GOSUB 2270
1590 DISPLAY AT(3,0):"OK! HAI VI
NTO.....", " ":" ":" " :: GOSUB 3
20
1600 R(1)=R(1)+1
1610 FOR I=1 TO 1000 :: NEXT I
1620 GOSUB 1910
1630 GOTO 720
1640 REM VITTORIA COMPUTER
1650 DISPLAY AT(3,0):"SPIACENTE!
IO MUOVO IN";X
1660 CAR=112
1670 GOSUB 1450
1680 JC=14 :: SC=2 :: GOSUB 2270
1690 DISPLAY AT(4,0):"E VINCO!!!
!" :: GOSUB 320
1700 FOR I=1 TO 1000 :: NEXT I
1710 R(2)=R(2)+1
1720 GOSUB 1910
1730 GOTO 540
1740 REM VALUTAZIONE
1750 IF S=3 AND TT=0 THEN L(X)=1
1760 IF S=3 AND TT=1 THEN T=T-10
00
1770 IF S=2 AND TT=1 THEN T=T-10

```

```

1780 IF TT=1 THEN 1810
1790 IF S=15 AND B=0 THEN 1640
1800 T=T+N(S+1)
1810 RETURN
1820 REM STAMPA
1830 JC=5 :: SC=8 :: GOSUB 2270
1840 DISPLAY AT(3,0):"TOCCA A ME
...": "MUOVO NELLA COLONNA.."
1850 DISPLAY AT(6,0):" "
1860 RETURN
1870 GOTO 1800
1880 CALL ERR(CD,TY,SE,LN)
1890 PRINT CD;TY;SE;LN
1900 RETURN NEXT
1910 REM DISEGNO GRIGLIA
1920 JC=9 :: SC=15
1930 K=1
1940 GOSUB 2270
1950 DISPLAY AT(1,10)ERASE ALL:"
# FILETTO #"
1960 DISPLAY AT(2,10):"=====
=="
1970 CALL CHAR(96,"000000FF")
1980 CALL CHAR(97,"3C3C3C3C3C3C
C3C")
1990 CALL CHAR(98,"FFFFFFFFFFFF
FFF")
2000 FOR I=9 TO 24 STEP 15
2010 CALL HCHAR(I,2,98,22)
2020 NEXT I
2030 FOR I=12 TO 21 STEP 3
2040 CALL HCHAR(I,2,96,22)
2050 NEXT I
2060 FOR I=2 TO 23 STEP 3
2070 CALL VCHAR(10,I,97,14)
2080 NEXT I
2090 FOR I=1 TO 19 STEP 3
2100 DISPLAY AT(8,I):(I+2)/3
2110 NEXT I
2120 FOR J=1 TO 7
2130 L(J),G(J)=0
2140 FOR I=1 TO 5
2150 A(I,J)=0
2160 NEXT I
2170 NEXT J
2180 C=0
2190 DISPLAY AT(9,23):"SCORE"
2200 DISPLAY AT(8,23):"====="
2210 DISPLAY AT(10,23):"====="
2220 DISPLAY AT(12,24):"TU"
2230 DISPLAY AT(14,24):R(1)
2240 DISPLAY AT(17,24):"IO"
2250 DISPLAY AT(19,24):R(2)
2260 RETURN
2270 REM COLORE STAMPA
2280 FOR I=0 TO 8
2290 CALL COLOR(I,JC,K)

```

2300 CALL SCREEN(SC)
2310 NEXT I

2320 RETURN



Pixelator

In quella parte di programma dove viene effettuata la definizione dei caratteri grafici, può capitare di definire un particolare carattere (o gruppo) per poi accorgersi che, una volta codificato in esadecimale (e dunque visualizzato sul video), non risulta quello che ci si aspettava effettivamente.

Questo fatto è dovuto alla riduzione alla quale il carattere codificato è soggetto ed al diverso contrasto che ogni coppia di colore possiede (ciò si accentua in modo particolare quando si devono trattare caratteri definiti molto piccoli). Può capitare inoltre che, una volta costruita una matrice e codificata in esadecimale, la si voglia cambiare leggermente ("accendendo" o "spegnendo" anche un singolo pixel).

Questo programma è stato creato per risolvere

re problemi attinenti alla grafica del Texas in TI BASIC. Una volta dato il RUN, appare sul video una matrice (8x8) sulla quale vi è un cursore bianco lampeggiante, che può essere spostato usando i soliti tasti con le frecce; la matrice appare inizialmente con tutti i suoi singoli elementi "spenti". Per fare "accendere" ogni singolo elemento, basta posizionarsi sopra di essi con il cursore, dopodiché va premuto "ENTER" (nel caso si volesse "spegnere" un certo elemento della matrice va eseguita la procedura di cui sopra).

Una volta che si è disegnata la sagoma desiderata sulla matrice basta premere "PROC'D" per avere sia la definizione "VERA" che quella "NEGATA" (ottenuta, cioè, cambiando i punti accesi con quelli spenti).

```
100 REM **#####**
110 REM * *
120 REM # PIXELATOR #
130 REM * *
140 REM **#####**
150 REM copyright: ALE
SSANDRO&ERNESTO&MASSIMO
160 REM per TEXAS TI-99/4A
170 REM DEFINIZIONE DEI CARATTE
RI :
180 CALL SCREEN(8)
190 CALL CLEAR
200 CALL CHAR(128,"00000000000000
001")
210 CALL CHAR(129,"00000000000000
0FF")
220 CALL CHAR(130,"00000000000000
080")
230 CALL CHAR(131,"8080808080808
080")
```

```
240 CALL CHAR(132,"80000000000000
000")
250 CALL CHAR(133,"FF000000000000
000")
260 CALL CHAR(134,"01000000000000
000")
270 CALL CHAR(135,"01010101010101
010")
280 CALL CHAR(136,"01010101010101
1FF")
290 CALL CHAR(137,"FF80808080808
080")
300 CALL HCHAR(3,6,128)
310 CALL HCHAR(3,7,129,6)
320 CALL HCHAR(3,13,130)
330 CALL VCHAR(4,13,131,6)
340 CALL HCHAR(10,13,132)
350 CALL HCHAR(10,7,133,6)
360 CALL HCHAR(10,6,134)
370 CALL VCHAR(4,6,135,6)
```

```

380 CALL HCHAR(6,3,128)
390 CALL HCHAR(6,4,129,2)
400 CALL HCHAR(6,6,136)
410 CALL HCHAR(6,7,129,3)
420 CALL HCHAR(6,10,130)
430 CALL VCHAR(7,10,131,3)
440 CALL HCHAR(10,10,137)
450 CALL VCHAR(11,10,131,2)
460 CALL HCHAR(13,10,132)
470 CALL HCHAR(13,4,133,6)
480 CALL HCHAR(13,3,134)
490 CALL VCHAR(7,3,135,6)
500 CALL HCHAR(8,7,65)
510 CALL HCHAR(8,8,38)
520 CALL HCHAR(8,9,69)
530 CALL HCHAR(9,8,77)
540 CALL HCHAR(11,5,83)
550 CALL HCHAR(11,6,79)
560 CALL HCHAR(11,7,70)
570 CALL HCHAR(11,8,84)
580 CALL HCHAR(5,8,49)
590 CALL HCHAR(5,9,57)
600 CALL HCHAR(5,10,56)
610 CALL HCHAR(5,11,52)
620 FOR DEL=1 TO 1.5E3
630 NEXT DEL
640 REM DEFINIZIONE COLORI
650 CALL CLEAR
660 CALL SCREEN(5)
670 FOR I=1 TO 12
680 CALL COLOR(I,2,8)
690 NEXT I
700 GOSUB 2170
710 PRINT "      INIZIALIZZAZIONE.
..":::::::::::::
720 OPTION BASE 0
730 DIM HEX$(15,1),MAT(7,7)
740 REM CARICO VETTORE DI
CONVERSIONE BIN-HEX
750 REM *****HEX$(
(N,0)=VAL. HEX. VERO      HEX$(N,1
)=VAL. HEX. NEGATO
760 FOR I=0 TO 15
770 READ HEX$(I,0),HEX$(I,1)
780 NEXT I
790 DATA 0,F,1,E,2,D,3,C,4,B,5,A
,6,9,7,8,8,7,9,6,A,5,B,4,C,3,D,2
,E,1,F,0
800 REM DEFINIZIONE GRAFICA
810 CALL CHAR(128,"0101010101010
1FF")
820 CALL CHAR(129,"0101010101010
101")
830 CALL CHAR(130,"0000000000000
0FF")
840 CALL CHAR(131,"FFFFFFFFFFFFF
FFF")
850 CALL CHAR(132,"00000000FF000
000")

```

```

860 CALL CHAR(133,"00010101FF010
101")
870 CALL COLOR(13,2,8)
880 CALL CHAR(136,"0101010101010
1FF")
890 CALL CHAR(137,"0000000000000
000")
900 CALL CHAR(138,"FFFFFFFFFFFFF
FFF")
910 CALL COLOR(14,2,16)
920 GOSUB 2170
930 REM DISEGNO TESTO
940 VTAB=1
950 HTAB=3
960 D$="      - * # PIXELATOR # * -
"
970 GOSUB 2210
980 VTAB=12
990 D$="S.VERA= 00000000000000000
=&CHR$(137)
1000 GOSUB 2210
1010 VTAB=14
1020 D$="S.NEG.= FFFFFFFFFFFFFFFF
F=&CHR$(138)
1030 GOSUB 2210
1040 VTAB=16
1050 D$="ENTER PER CAMBIARE IL P
UNTO"
1060 GOSUB 2210
1070 VTAB=18
1080 D$="PROC'D PER CALCOLARE"
1090 GOSUB 2210
1100 VTAB=20
1110 D$="ERASE PER CANCELLARE"
1120 GOSUB 2210
1130 VTAB=22
1140 D$="D,S,X,E PER MUOVERE"
1150 GOSUB 2210
1160 VTAB=24
1170 D$="QUIT PER FINIRE"
1180 GOSUB 2210
1190 CALL HCHAR(3,18,137)
1200 CALL HCHAR(10,18,138)
1210 CALL HCHAR(5,16,137,3)
1220 CALL HCHAR(8,16,138,3)
1230 FOR I=20 TO 22
1240 CALL VCHAR(3,I,137,3)
1250 CALL VCHAR(8,I,138,3)
1260 NEXT I
1270 CALL VCHAR(3,24,137,3)
1280 CALL VCHAR(8,24,138,3)
1290 FOR I=12 TO 26 STEP 2
1300 CALL HCHAR(13,I,132)
1310 CALL HCHAR(13,I+1,133)
1320 NEXT I
1330 REM DISEGNO CARATTERE
1340 GOSUB 2260
1350 CALL KEY(3,K,S)

```

```

1360 IF S=0 THEN 1480
1370 IF K=68 THEN 1560
1380 IF K=83 THEN 1610
1390 IF K=88 THEN 1660
1400 IF K=69 THEN 1710
1410 IF K=13 THEN 1760
1420 IF K=12 THEN 1870
1430 IF K=7 THEN 1340
1440 FOR LIP=0 TO 30 STEP 2
1450 CALL SOUND(-200,-5,LIP)
1460 NEXT LIP
1470 GOTO 1350
1480 REM CURSORE
1490 CALL HCHAR(YC+3,XC+6,136)
1500 FOR D=1 TO 4
1510 NEXT D
1520 CALL HCHAR(YC+3,XC+6,128+3*
MAT(YC,XC))
1530 FOR DELAY=1 TO 4
1540 NEXT DELAY
1550 GOTO 1350
1560 REM CURSORE A DESTRA
1570 XC=XC+1
1580 IF XC<=7 THEN 1480
1590 XC=0
1600 GOTO 1480
1610 REM CURSORE A SINISTRA
1620 XC=XC-1
1630 IF XC>=0 THEN 1480
1640 XC=7
1650 GOTO 1480
1660 REM CURSORE IN BASSO
1670 YC=YC+1
1680 IF YC<=7 THEN 1480
1690 YC=0
1700 GOTO 1480
1710 REM CURSORE IN ALTO
1720 YC=YC-1
1730 IF YC>=0 THEN 1480
1740 YC=7
1750 GOTO 1480
1760 REM ENTER
1770 CALL SOUND(100,440,4)
1780 IF MAT(YC,XC)=0 THEN 1830
1790 MAT(YC,XC)=0
1800 CALL HCHAR(YC+3,XC+6,128)
1810 CALL KEY(3,K,S)
1820 IF S<0 THEN 1810 ELSE 1480

1830 MAT(YC,XC)=1
1840 CALL HCHAR(YC+3,XC+6,131)
1850 CALL KEY(3,K,S)
1860 IF S<0 THEN 1850 ELSE 1480

1870 REM CALCOLO : STRINGA CHV
$= S. VERA CHN$= S. NEG
1880 CHV$=""
1890 CHN$=""
1900 FOR YC=0 TO 7
1910 VN=0
1920 FOR XC=0 TO 3
1930 VN=VN+MAT(YC,XC)*2^(3-XC)
1940 NEXT XC
1950 CHV$=CHV$&HEX$(VN,0)
1960 CHN$=CHN$&HEX$(VN,1)
1970 VN=0
1980 FOR XC=4 TO 7
1990 VN=VN+MAT(YC,XC)*2^(7-XC)
2000 NEXT XC
2010 CHV$=CHV$&HEX$(VN,0)
2020 CHN$=CHN$&HEX$(VN,1)
2030 NEXT YC
2040 HTAB=11
2050 VTAB=12
2060 CALL CHAR(137,CHV$)
2070 D$=CHV$
2080 GOSUB 2210
2090 VTAB=14
2100 CALL CHAR(138,CHN$)
2110 D$=CHN$
2120 GOSUB 2210
2130 YC=0
2140 XC=0
2150 GOTO 1480
2160 END
2170 REM CALL VCLEAR
2180 CALL VCHAR(1,31,31,96)
2190 CALL VCHAR(1,3,32,672)
2200 RETURN
2210 REM DISPLAY AT
2220 FOR CH=1 TO LEN(D$)
2230 CALL HCHAR(VTAB,HTAB-1+CH,A
SC(SEG$(D$,CH,1)))
2240 NEXT CH
2250 RETURN
2260 REM INIZIALIZZAZIONE
E STAMPA MATRICE
2270 FOR YC=0 TO 7
2280 FOR XC=0 TO 7
2290 MAT(YC,XC)=0
2300 NEXT XC
2310 NEXT YC
2320 YC=0
2330 XC=0
2340 REM STAMPA
2350 CALL CHAR(137,"000000000000
0000")
2360 CALL CHAR(138,"FFFFFFFFFFFF
FFFF")
2370 CALL HCHAR(12,11,48,16)
2380 CALL HCHAR(14,11,70,16)
2390 FOR R=3 TO 10
2400 CALL HCHAR(R,6,128,8)
2410 NEXT R
2420 CALL VCHAR(3,5,129,8)
2430 CALL HCHAR(2,6,130,8)
2440 RETURN
2450 END

```

Numeri in scala

Lo scopo nascosto di questo gioco è di verificare la vostra capacità decisionale. Lo Spectrum vi proporrà 10 numeri casuali (naturalmente uno alla volta) compresi tra zero e novantanove. Via via dovrete decidere se il numero in quel momento considerato va o no inserito fra i gradini di una scala colorata, posta sulla sinistra del video.

I parametri per la decisione riguardano il valore assoluto del numero e i posti ancora liberi sulla scala. Infatti più il numero è alto, più deve essere collocato in alto sulla scala. Segue da ciò che se nel nono gradino (il più alto) avete già posto il numero 85, tutti i numeri superiori a tale valore non potranno essere inseriti sulla scala. Ogni volta che rifiuterete un numero toglierete un punto dal vostro totale (inizialmente 10); ne aggiungerete uno per ogni numero correttamente posizionato.

```

1 INK 1: GO TO 1000
2 LET a=5: LET j=10
3 LET h$="corretto": LET max=
  100: LET min=0
4 DIM b(10): DIM e(10): DIM w
  (10)
5 FOR f=1 TO 10: LET e(f)=0:
  LET w(f)=100: NEXT f
6 BORDER 7: PAPER 7: CLS
7 PRINT AT 0,a;" {2SG8}"
8 FOR m=1 TO 2
9 FOR v=0 TO 19
10 POKE 23609,100
11 PRINT AT v,a-1;"{SG5}"
12 NEXT v
13 LET a=12
14 LET v=0
15 NEXT m
18 LET a=5
19 LET l=1: FOR n=1 TO 9:
  PRINT AT 1,a-1;CHR$ 142;"
  {6SG3}": LET l=l+2: NEXT n
20 INK 2
21 PRINT AT 2,a;" {2SG8}"
22 INK 3
23 LET w=9: LET q=0: FOR u=1

```

Nel caso abbiate impostato un posizionamento scorretto, vi potrete trovare in due situazioni: o avete cercato di inserirlo in un gradino già occupato o avete posto un valore fuori ordine rispetto agli altri. In entrambi i casi avete due possibilità di uscita: il tasto 'R' per ripetere l'operazione, se avete sbagliato nel premere un tasto e siete in grado di collocare il numero correttamente; il tasto 'E' se avevate proprio giudicato male e non è possibile inserire il numero sulla scala senza contravvenire alle regole del gioco: in questo caso perderete due punti.

Se volete conseguire un alto punteggio dovete sviluppare una strategia di lungo periodo; cercando di limitare il più possibile il numero di rifiuti: il massimo punteggio realizzabile è 20, ma noi non conosciamo nessuno che ci sia mai riuscito!!

```

  TO 10: PRINT AT q,a-3;w;"
  ": LET q=q+2: LET w=w-1:
  NEXT u
30 PRINT AT 4,a;" {2SG8}"
31 INK 4
40 PRINT AT 6,a;" {2SG8}"
41 INK 5
50 PRINT AT 8,a;" {2SG8}"
51 INK 6
60 PRINT AT 10,a;" {2SG8}"
61 INK 0
70 PRINT AT 12,a;" {2SG8}"
71 INK 1
80 PRINT AT 14,a;" {2SG8}"
81 INK 2
90 PRINT AT 16,a;" {2SG8}"
91 INK 3
100 PRINT AT 18,a;" {2SG8}"
200 REM
210 LET spazio=0
220 LET colonna=0
230 LET contrasto=0
240 LET attributo=0
300 REM
301 LET punti=10
305 FOR x=0 TO 9

```

```

310 LET b(j)=INT (RND*99)
312 FOR f=1 TO 10: IF b(j)=w(f)
  THEN GO TO 310: NEXT f
315 PRINT AT 0,20; FLASH 1;
  INK 3;"PUNTI="; FLASH 0;pu
  nti;" "
320 PRINT AT 20,0;"
  "; FLASH 1; INK 0;b(j)
  ;" S / N ?"; INVERSE 0;
  FLASH 0;"
"
330 LET b$=INKEY$
340 IF b$="n" THEN LET punti=p
  unti-1: GO TO 600
350 IF b$="s" THEN GO TO 400
360 GO TO 330
400 REM
405 LET h$="corretto": LET max=
  100: LET min=0
410 INK 0: PRINT AT 20,0;"IN QU
  ALE BLOCCO VUOI POSIZIONARE
  IL NUMERO ";b(j);" ( 0 - 9
  ) ? "
415 INK 2
420 INPUT a$
430 LET d=VAL a$
435 GO SUB 3000
440 PRINT AT 21,0;"
"
445 IF h$="" THEN LET h$="r"
450 IF h$="corretto" THEN
  GO TO 500
460 IF h$="e" THEN LET punti=p
  unti-2: GO TO 600
470 IF h$="r" THEN GO TO 400
500 REM
510 LET t$=STR$ b(j)
520 LET len=LEN t$
530 IF len=2 THEN LET spazio=2
"
540 IF len=1 THEN LET spazio=3
"
550 LET linea=18-d*2: LET colon
  na=a+spazio
560 LET attributo=ATTR (linea,c
  olonna)
565 LET contrasto=INT ((attribu
  to)/8)
566 LET sfondo=attributo-contras
  to*8
570 PRINT AT linea,colonna;
  PAPER sfondo; INK contrast
  o;;b(j)
580 LET punti=punti+1
585 LET w(d+1)=b(j)
590 LET e(d+1)=1
600 NEXT x
700 REM punteggio finale

```

```

710 INK 0: PAPER 4: BORDER 3:
  CLS : FOR v=1 TO 20:
  BEEP .1,v: NEXT v
720 CLS : FLASH 1: PRINT AT 10,
  6;"Il tuo punteggio e' ";pu
  nti
730 IF punti>=12 THEN PRINT
  AT 12,8;"CONGRATULAZIONI !!
  !": FOR g=1 TO 20: BEEP .5,
  g: NEXT g
740 IF punti>=15 THEN PRINT
  AT 14,6;"SEI STATO BRAVISSI
  MO !!": FOR g=1 TO 10:
  BEEP .5,g: NEXT g
750 IF punti=20 THEN PRINT
  AT 16,3;"e anche un po' FOR
  TUNATO !!!": FOR g=1 TO 5:
  BEEP .5,g: NEXT g
760 FLASH 0
770 INK 0: PRINT AT 20,5;"Vuoi
  provare ancora ? ": IF
  INKEY$="n" THEN STOP
780 PRINT AT 21,5;"Premi < S >
  o < N >": IF INKEY$="n"
  THEN STOP
790 IF INKEY$="s" THEN RUN 2
800 GO TO 770
1000 PAPER 6: INK 0: BORDER 2:
  CLS : PRINT AT 10,6;
  FLASH 1;"NUMERI SULLA SCAL
  A "
1001 FOR y=1 TO 10
1002 FOR o=1 TO 3
1003 BEEP .1,y
1004 BEEP .1,o
1005 NEXT o: NEXT y
1006 BEEP 3,11
1007 PAPER 7
1008 CLS
1009 INK 2
1010 PRINT AT 0,7;"*** ISTRUZION
  I ***"
1020 PRINT
1030 INK 0: PRINT "Lo Spectrum s
  cegliera' 10 numeriche devi
  cercare di posizionare sec
  ondo il loro valore sui
  gradini di una scala (piu
  ' il numero e' alto piu'
  deve essere alto il gradino
  )."
1035 PRINT ""Quando appare il nu
  mero, devi decidere, in
  base alla tua strategia
  ,se scartarlo e perdere un p
  unto oppure se inserirlo,
  condizionando pero' cosi'
  le scelte future!"

```

```

1036 PRINT "Parti con 10 punti:
        cerca di non perderne tropp
        i....Buona Fortuna"
1040 PRINT
1045 FLASH 1
1050 PRINT AT 20,3;"Premi un tas
        to per iniziare"
2000 PAUSE 0
2010 FLASH 0
2100 GO TO 2
3000 REM
3010 IF e(d+1)=1 THEN BEEP .03,
        5: BEEP .02,10: PRINT AT 20
        ,0; FLASH 1;"      POSIZIONAM
        ENTO SCORRETTO      ";
        FLASH 0;"<E> per uscire, <
        R> per ripetere": INPUT h$:
        RETURN

```

```

4000 REM
4010 FOR f=d+1 TO 9
4020 IF e(f+1)=1 THEN LET max=w
        (f+1): LET f=9
4030 NEXT f
4040 FOR f=d-1 TO 0 STEP -1
4050 IF e(f+1)=1 THEN LET min=w
        (f+1): LET f=0
4060 NEXT f
4100 IF b(j)>max OR b(j)<min
        THEN BEEP .03,5: BEEP .02
        ,10: PRINT AT 20,0; FLASH 1
        ;"      POSIZIONAMENTO SCORRE
        TTO      "; FLASH 0;"<E> per
        uscire, <R> per ripetere":
        INPUT h$: RETURN
4500 RETURN

```

Cubi spaziali



16/48K

Gli abitanti del Pianeta nero sono in pericolo: misteriosi cubi spaziali rotolano verso il loro pianeta distruggendo tutto ciò che incontrano nel loro cammino minacciando così la loro stessa esistenza.

Voi, armati di potenti missili interstellari, ten-

terete di arrestare l'avanzata dei terribili nemici cercando inoltre di evitare i raggi laser che tempestano la loro base.

I comandi sono molto semplici: "1" per muoversi verso sinistra "2" per muoversi verso destra "5" per sparare.

```

1 BORDER 0: PAPER 0: INK 7:
  LET sc=0: GO SUB 300:
  LET p=.1
2 CLS : LET shs=6: LET p=p/1.
  3: IF p<0.0007 THEN LET p=
  0.0007
3 PRINT AT 0,0;"QQQQQQQQQQQQQQQQ
QQQQQQQQQQQQQQQQQQQQQQ"
4 PRINT AT 1,0;CHR$ 137;
  CHR$ 140;CHR$ 134: PLOT
  OVER 1;10,162: PLOT
  OVER 1;14,162
5 FOR j=1 TO 100
6 PLOT INK RND*8;RND*255,
  RND*175
7 NEXT j
8 FOR q=0 TO 2: LET t=(RND*21
  0)+20: LET l=(RND+130)+20:
  FOR h=1 TO (RND*4)+3:

```

```

  CIRCLE t,1,h: NEXT h:
  NEXT q
9 FOR r=2 TO 15: BEEP .3,r:
  NEXT r: PLOT 0,0: DRAW 255
  ,0,-1.2
20 LET i=22593
21 LET n=1
40 LET x=22528+703-15
45 PAUSE 50
50 POKE x,56: POKE x-1,0:
  POKE x+1,0
51 PRINT INVERSE 1;AT 0,0;"PU
  NTI ";sc;AT 0,23;"MISSILI "
  ;shs
55 IF n=703 THEN BEEP 3,23:
  POKE 23609,0: PRINT AT 12,
  0;"IL CUBO SPAZIALE TI HA R
  AGGIUNTO""Hai totalizzato
  ";sc;" punti.": FOR q=1

```



```

    TO 250: NEXT q: GO TO 900
60 POKE i+n,56: POKE i+n-1,0:
    BEEP p,20
62 IF RND>.9 THEN GO SUB 500
65 IF IN 63486=255 THEN LET n
    =n+1: POKE 23618,50: POKE 2
    3620,1
70 IF IN 63486=190 OR IN 63486
    =254 THEN LET x=x-1: LET n
    =n+1: POKE 23618,50: POKE 2
    3620,1
80 IF IN 63486=189 OR IN 63486
    =253 THEN LET x=x+1: LET n
    =n+1: POKE 23618,50: POKE 2
    3620,1
90 IF IN 63486=175 OR IN 63486
    =239 THEN LET shs=shs-1:
    POKE 23618,150: POKE 23620
    ,1
100 LET n=n+1: POKE 23618,50:
    POKE 23620,1
150 IF shs<=0 THEN LET shs=0:
    POKE 23618,100: POKE 23620
    ,1
151 FOR k=0 TO 31
160 IF ATTR (21,k)=56 THEN
    GO TO 180
170 NEXT k
175 POKE 23618,100: POKE 23620,
    1
180 FOR f=19 TO 0 STEP -1
190 PRINT AT f,k;"A";AT f+1,k;"
    "
192 BEEP .02,f
195 IF ATTR (f-1,k)=56 THEN
    LET sc=sc+shs*10: LET shs=
    5: PRINT AT f-1,k;" ";AT f,
    k;" ": PAUSE 70: LET n=1:
    BEEP 2,-10: POKE 23618,50:
    POKE 23620,1
197 NEXT f
200 POKE 23618,100: POKE 23620,
    1
300 FOR g=0 TO 7
305 READ s
310 POKE USR "A"+g,s
320 NEXT g
330 DATA 24,24,24,24,24,126,90,
    129
335 RESTORE 400
340 FOR g=0 TO 7: READ s
350 POKE USR "Q"+g,s: NEXT g
360 DATA 34,62,28,28,8,8,62,42
400 DATA 24,62,28,28,8,8,62,42
410 RETURN
500 LET k=RND*255: PLOT k,170:
    DRAW 0,-160
505 FOR m=0 TO 30: NEXT m:

```

```

    DRAW OVER 1;0,159
510 IF ATTR (21,INT (k/8))=56
    THEN GO TO 800
515 RETURN
800 PRINT AT 12,0;"SEI STATO CO
    LPITO!""Hai totalizzato "
    ;sc;" punti.": FOR q=1 TO 2
    50: NEXT q: GO TO 900
900 CLS : INPUT "VUOI GIOCARE A
    NCORA(s/n)";a$
910 IF a$="s" OR a$="S" THEN
    RUN
920 STOP

```

Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate —, dovrete battere CONTROL+SPAZIO, e così via.

ERRATA CORRIGE

Nel programma DEMON STAR versione per VIC 20 su Papersoft n. 14, le linee 62 e 63 del programma I appaiono in fondo al listato stesso. Occorre inoltre eliminare l'ultima parte della linea 61:

```
:GOSUB75:RG=5:GOTO65
```

ed aggiungerla in coda alla linea 63, che dovrà quindi diventare:

```
63 GOSUB70:IFRP<7706THENGOSUB70:
    POKERP,47:POKERC,6:POKEVN,241
    :GOSUB75:RG=5:GOTO65
```

Il programma I finisce quindi con la linea 75.

Istogrammi a barre

'Istogrammi a barre' è un buon esempio di ciò che i programmatori alla prime armi possono ottenere in fatto di grafica sul Commodore 64. Per realizzare questo programma sono state utilizzate diverse tecniche grafiche.

'Istogrammi a barre' crea attraenti grafici a barre con colonne tridimensionali. I grafici so-

no particolarmente interessanti per ciò che riguarda i colori. Sono disponibili sei colonne di diverso colore, a cui si può assegnare un valore da 0 a 15 per l'altezza della colonna. Potete fornire al calcolatore il valore assegnato a ciascuna colonna, e quindi interpretarne il significato.

```
100 PRINT "{CLR}":POKE53281,0:POKE53280,6
110 FORI=1230TO1261:POKEI,114:POKEI+54272,5:NEXT:REM ALTO
120 FORI=1270TO1790STEP40:POKEI,107:POKEI+54272,5:NEXT:REM SINISTRA
130 FORI=1301TO1821STEP40:POKEI,115:POKEI+54272,5:NEXT:REM DESTRA
140 FORI=1830TO1861:POKEI,113:POKEI+54272,5:NEXT:REM BASSO
150 FORI=1351TO1380:POKEI,67:POKEI+54272,5:NEXT
160 FORI=1471TO1500:POKEI,67:POKEI+54272,5:NEXT
170 FORI=1591TO1620:POKEI,67:POKEI+54272,5:NEXT
180 FORI=1711TO1740:POKEI,67:POKEI+54272,5:NEXT:PRINT:PRINT:PRINT
190 PRINT:PRINTSPC(4)"15":PRINTSPC(4)"14":PRINTSPC(4)"11":PRINTSPC(4)"12"
200 PRINTSPC(4)"11":PRINTSPC(4)"10":PRINTSPC(5)"9":PRINTSPC(5)"8"
210 PRINTSPC(5)"7":PRINTSPC(5)"6":PRINTSPC(5)"5":PRINTSPC(5)"4"
220 PRINTSPC(5)"3":PRINTSPC(5)"2":PRINTSPC(5)"1":PRINTSPC(5)"0"
230 PRINTTAB(9)"A"SPC(4)"B"SPC(4)"C"SPC(4)"D"SPC(4)"E"SPC(4)"F"
240 IFZ$="S"THEN310
250 DATAA,B,C,D,E,F
260 FORI=1TO6:READA$(I):NEXTI
```

```
270 DATA7,6,4,13,8,14
280 FORI=1TO6:READD(I):NEXTI
290 DATA1792,1797,1802,1807,1812,1817
300 FORI=1TO6:READA(I):NEXTI
310 C=54272:FORJ=1TO6
320 FORK=1971TO1975:POKEK,32:NEXTK:PRINT"{HOME}":FORL=1TO22:PRINT:NEXTL
330 PRINT"VALORE PER LA COLONNA";A$(J);"(0-15)";
340 INPUTA$:IFVAL(A$)=0THEN320
350 AA=VAL(A$):AA=INT(AA+.5):D=D(J):X=A(J)
360 IFAA<0ORAA>15THEN320
370 A=A(J)-(AA*40)+80:GOSUB430:NEXTJ
380 PRINT"{HOME}":FORI=1TO22:PRINT:NEXT:FORI=1971TO1983:POKEI,32:NEXT
390 PRINT"VUOI FARNE UN ALTRO (S/N){ 3 SPAZI}"
400 GETZ$:IFZ$=""THEN400
410 IFZ$="S"THEN100
420 END
430 IFAA=1THEN530
440 IFAA=0THENRETURN
450 POKEX,160:POKEX+1,231:POKEX+2,105
460 POKEX+C,D:POKE(X+1)+C,D:POKE(X+2)+C,D:IFAA=2THEN490
470 FORI=X-40TOASTEP-40:POKEI,160:POKEI+1,231:POKEI+2,160
480 POKEI+C,D:POKE(I+1)+C,D:POKE(I+2)+C,D:NEXTI:GOTO510
490 POKEA,247:POKEA+1,208:POKEA+2,105
500 POKEA+C,D:POKE(A+1)+C,D:POKE(A+2)+C,D:GOTO530
```

```

510 POKEA,247:POKEA+1,208:POKEA+
    2,224
520 POKEA+C,D:POKE(A+1)+C,D:POKE
    (A+2)+C,D
530 POKEA-40,233:POKEA-39,160:PO

```

```

    KEA-38,206
540 POKE(A-40)+C,D:POKE(A-39)+C,
    D:POKE(A-38)+C,D
550 RETURN

```



SuperSprite

SuperSprite, un essere dotato di poteri soprannaturali, deve volare attraverso quattro barriere di kryptonite dotate solo di stretti passaggi; fortunatamente voi potete aiutarlo: muovendo accortamente attraverso i tasti di funzione le barriere stesse dovete cercare di fargli attraversare indenne lo schermo facendo in modo che nel suo imprevedibile volo non incontri mai le mortali barriere.

Avete 20 manches a disposizione e se SuperSprite urta una barriera perdete una manche e il vostro amico ricomincerà il suo folle volo dall'alto dello schermo.

Il programma memorizza il massimo punteggio realizzato all'interno del LIST stesso e quindi, dopo aver battuto il programma, prima di salvarlo la prima volta dovete battere POKE PEEK(44)x256+PEEK(43)+5,1 per inizializzare il vostro primo massimo punteggio: cioè 0!

Se battete il record, un messaggio vi ricorda di salvare il programma per memorizzare il nuovo record assoluto.

Le istruzioni dettagliate del gioco sono inserite nel programma stesso: non dovete fare altro che inserirlo e... a voi!

```

110 U=RND(-TI):POKE657,128
120 REM MEMORIZZAZIONE PREC.SITU
    AZIONE,I          NIZIALIZZAZI
    ONE NUOVA,COSTRUZIO
125 REM NE SPRITE,STAMPAGGIO IST
    RUZIONI
130 SP=13:V=53248
140 POKE2041,PEEK(V+24):POKEV+24
    ,22
150 POKE2042,PEEK(V+32):POKEV+32
    ,7
160 POKE2043,PEEK(V+33):POKEV+33
    ,7
170 POKE2044,PEEK(646):POKE646,0
180 POKE2045,PEEK(650):POKE650,1
    28
190 PRINT"{CLR}":GOSUB4000
200 GOSUB 6000:PRINT"{CLR}":POKE
    V+24,20
210 DEFNSC(U)=INT(U+U*GL/10)
220 FORI=1TO16:T$=T$+"{DES}":NEX
    TI

```

```

230 B$="{ 3 GIU' }"+CHR$(13)+"
    {RVS}":FORI=0TO39:B$=B$+" ":
    NEXTI
240 V=53248:C=55296:S=1024:SD=54
    272
250 HS=PEEK(43)+PEEK(44)*256+5
260 T1=135:T2=255:T3=230:T4=133
270 T5=5:T6=9:T7=.992:T8=42:T9=3
    2
280 BS=PEEK(HS):CH=65508:AR=780
290 REM INIZIALIZ. CURSORI,STAMP
    A SCHERM        O DI GIOCO
300 E(1)=1269:E(2)=1475:E(3)=168
    2:E(4)=1888
310 POKEV+21,0:GOSUB2000
320 REM CREAZIONE SPAZIO SULLO S
    CHERMO P        ER VISUALIZZ
    AZIONE PUNTEGGIO
325 REM INIZIALIZ. SID,AZZERAMEN
    TO DEL BU        FFER DI TAST
    IERA
330 FORI=1080TO1100:POKEI,32:NEX
    T

```

```

340 POKESD+4,16:POKESD+11,16:POK
ESD+18,16
350 POKESD+6,240:POKESD+13,240:P
OKESD+20,240
360 POKESD+24,15
370 GETAŞ:IFAŞ<>" "THEN370
380 :
390 REM INIZIO GIOCO(20 MANCHES)
,PUNTEG GIO PARZIALE
A ZERO
400 FORGL=1TO20:PS=0
410 REM POSIZIONAMENTO SPRITE,ZE
RO NEL R EGISTRO RILE
VAMENTO COLLISIONE
420 POKEV,23:POKEV+1,49
430 POKEV+23,0:POKEV+31,0
440 REM STAMPA PUNTEGGI,INIZIO S
UONO
450 PRINT"{HOME}{GIU' }"T$ "MANCHE
:"GL" PUNTEGGIO:"TS
460 PRINT"{GIU' }"T$ "{ 2 DES}MIGL
IOR RISULTATO:"PEEK(HS)-1
470 POKESD+4,17:POKESD+11,17:POK
ESD+18,17
480 REM ABILITAZIONE SPRITE,VAI
A 570 SE IL GIOCO E'
FINITO(FQ SETTATO)
490 POKEV+21,1:GOSUB1000:IFFQ=1T
HEN550
500 REM SPRITE FUORI DA SCHERMO
SE NON C' E' COLLISION
E,ALTRIMENTI...CRASH!
510 IFPEEK(V+1)>T3THENFORI=PEEK(
V+1)TO250:POKEV+1,I:NEXTI:GO
TO550
520 POKESD+4,129:POKESD+11,33:PO
KESD+18,33
530 FORI=72TO20STEP-2:POKESD+1,I
:POKESD+8,I-3:POKESD+15,I-2:
NEXTI
540 REM NO SUONO.SALTA A USCITA
SE FQ E'1
550 POKESD+4,16:POKESD+11,16:POK
ESD+18,16
560 POKESD+1,0:POKESD+8,0:POKESD
+15,0
570 IFFQ=1THENFQ=0:GL=20:NEXT:PO
KEV+21,0:GOTO670
580 REM CALCOLO PUNTEGGIO PARZIA
LE DALL'O RDINATA SPRI
TE,MODIFICA,AGGIUNTA
585 REM AL PUNTEGGIO TOTALE
590 PS=FNSC(INT((PEEK(V+1)-71)/4
0))
600 TS=TS+INT((PS↑1.4+PS)/2)
610 POKEV+21,0:POKEV+16,0
630 REM AGGIORNAMENTO PUNTEGGIO
RECORD, AZZERAMENTO
SID,ESCI(3000)
640 IFPEEK(HS)<TS+1THENPOKEHS,TS
+1
650 FORI=0TO24:POKESD+I,0:NEXTI
660 FORI=1TO500:NEXTI
670 GOSUB3000:CLR:GOTO210
680 :
970 REM SOTTOPROGRAMMA---UNA MAN
CHE DEL GIOCO
980 :
990 REM DECIDI QUALE REGISTRO PO
SIZ.SPRIT E CAMBIARE(W
) E DI QUANTO(U)
1000 U2=RND(1):W=V+INT(U2*2):U=G
L/7+2
1010 REM ESPANDI SPRITE IN DIR.Y
SE E' VERA L'ESP
RESSIONE
1020 IFU2>T7ANDPEEK(V+1)<T1THENP
OKEV+23,1
1030 REM SETTA L'OTTAVO BIT DELL
A POSIZIO NE X DELLO
SPRITE SE NECESSARIO
1040 IFPEEK(W)+U>T2THENPOKEV+16,
1:POKEV,PEEK(V)+U-T2:GOTO10
70
1050 REM AGGIORNA SPRITE.ESCI SE
COLLISI ONE O BARR
IERE SUPERATE
1060 POKEW,PEEK(W)+U
1070 IF(PEEK(V+31)AND1)ORPEEK(V+
1)>T3THENRETURN
1080 REM CAMBIO FREQUENZE SECOND
O POSIZ.Y DELLO SPRI
TE E NUMERO CASUALE U2
1090 U=232-PEEK(V+1):POKESD+1,U*
2/3:POKESD+8,U+3*U2:POKESD+
15,U+3
1100 REM SCANDISCI TASTIERA:ESCI
SE *;PAUSA SE' ';VAI A 100
0 SE NON E'TASTO FUNZ
1110 SYSCH:U=PEEK(AR):IFU=T8THEN
TS=0:FQ=1:RETURN
:REM USCITA
1120 IFU=T9THENGOSUB8000
1130 IFU<T4THEN1000:REMINPUT<F1
1140 REM MOVIMENTO LINEA:DX SE N
ON SCIFTATO(1120):SIN SE SC
IFTATO(1140)
1150 U=U-132:IFU<T5THEN1180:REMI
NPUTF1-F4
1160 IFU<T6THEN1200:REM INPUT F5
-F8
1170 GOTO1000:REM INPUT>F8
1180 E(U)=E(U)+(E(U)=1100+200*U-
(U>2)):POKEE(U),223:POKEE(U
)-1,160
1190 POKEE(U)+4+(U>2),32:POKEE(U
)+5+(U>2),95:E(U)=E(U)+1:GO
TO1000

```

```

1200 U=U-4:E(U)=E(U)-(E(U)=1064+
200*(U)):POKEE(U)+3+(U>2),9
5
1210 POKEE(U)+4+(U>2),160:POKEE(
U)-1,32:POKEE(U)-2,223:E(U)
=E(U)-1:GOTO1000
1970 REM SOTTOPROGRAMMA---STAMPA
SCHERMO

1980 :
1990 REM I COLORI DELLE LINEE AL
LA RIGA 2000:L'ULT
IMO PER LE SCRITTE
2000 PRINT"{CLR}{ 2 GIU' }{PUR}"B
$"{GRN}"B$"[<7>]"B$"[<3>]"B
$"{HOME}{BLK}"
2010 FORI=1TO4:POKEE(I)-1,223:PO
KEE(I)+3-(I<3),95:FORJ=E(I)
TOE(I)+2-(I<3)
2020 POKEJ,32:NEXTJ,I:RETURN
2030 :
2980 REM SOTTOPROGRAMMA--USCITA
O RIENTRO
2990 :
3000 PRINT"{CLR}{GIU' }"T$"
{ 3 SPAZI}PUNTEGGIO:"TS
3010 PRINT"{GIU' }"T$"{ 6 SIN} MI
GLIOR RISULTATO:";PEEK(HS)-
1
3020 PRINT"{ 3 GIU' }GIOCHI ANCOR
A (S/N)?
3030 GETA$:IFA$=""THEN3030
3040 IFA$="S"THENRETURN
3050 IFA$="N"THEN3070
3060 GOTO3030
3070 IFNOT(BS<PEEK(HS))THEN3100
3080 PRINT:PRINT"CONGRATULAZIONI
, HAI BATTUTO IL RECORD.
3090 PRINT:PRINT"RICORDA DI SALV
ARE IL PROGRAMMA.
3100 PRINT:PRINT"ARRIVEDERCI A P
RESTO!
3110 PRINT"{ 4 GIU' }"
3120 REM RITORNO CONDIZIONI INIZ
IALI
3130 POKEV+24,PEEK(2041):POKEV+3
2,PEEK(2042):POKEV+33,PEEK(
2043)
3140 POKE646,PEEK(2044):POKE650,
PEEK(2045):POKE657,0
3150 END
3160 :
3980 REM SOTTOPROGRAMMA--COSTRUZ
IONE SPRITE A PAG.13 (832-8
95 NEL BUFFER NASTRO)
3990 :
4000 FORI=0TO41:READU:POKE(64*SP
)+I,U:NEXT
4010 FORI=42TO63:POKE(64*SP)+I,0

```

```

:NEXT
4020 POKE2040,SP:POKEV+23,1:POKE
V+29,1:POKEV+39,0:RETURN
4030 RETURN
4040 DATA 248,0,0,62,0,0,7,128,0
,1,224
4050 DATA 0,0,120,0,0,62,0,0,15,
112
4060 DATA 0,7,248,0,7,248,0,6,12
0,0
4070 DATA 6,12,0,6,6,0,3,6,0,1,1
28
4080 :
4980 REM FRASI DATA PER PAGINA I
STRUZIONI
4990 :
5000 DATA"DEVI MANOVRARE LE LINE
E ORIZZONTALI
5010 DATA"IN MODO CHE SUPERSPRIT
E POSSA VOLARE
5020 DATA"SENZA DANNO ATTRAVERSO
I PASSAGGI.
5030 DATA"LE QUATTRO LINEE SONO
CONTROLLATE DAI
5040 DATA"TASTI DI FUNZIONE SULL
A TUA DESTRA.
5060 DATA"PREMENDO UNO DI ESSI F
ARAI SCORRERE
5070 DATA"LA CORRISPONDENTE LINE
A VERSO DESTRA.
5080 DATA"LO STESSO TASTO SHIFTA
TO LA SPOSTERA'
5090 DATA"VERSO SINISTRA.
{ 10 SPAZI}PREMI SPAZIO
5100 DATA"PER ARRESTARE IL GIOCO
, '*' PER FINIRE.
5110 DATA"
5120 DATA"UNA PARTITA CONSISTE D
I 20 MANCHES.
5130 DATA"IL PUNTI SI CONQUISTANO
ATTRAVERSANDO LE
5135 DATA"LINEE SENZA TOCCARLE.

```

PAPER
18 soft

ritaglia
e
conserva
ti
attendono
grosse
sorpresa!

```

5140 DATA" I PUNTEGGI AUMENTANO C
      ON IL NUMERO DI
5150 DATA" LINEE ATTRAVERSATE SEN
      ZA DANNO E CON IL
5160 DATA" NUMERO DI MANCHE FATTE
      . COL PROCEDERE
5170 DATA" DEL GIOCO AUMENTA SIA
      LA VELOCITA' CHE
5180 DATA" L'IRREGOLARITA' DEL VO
      LO DI SUPERSPRITE
5200 DATA" IL MASSIMO PUNTEGGIO P
      OSSIBILE E' 253.
5210 DATA" { 11 SPAZI } BUONA FORTU
      NA!
5215 DATA"
5220 DATA" { 4 SPAZI } PREMI { SPAZI }
      SPAZIO { SPAZI } PER { SPAZI } COM
      INCIARE.
5230 :
5980 REM SOTTOPROGRAMMA-STAMPA I
      STRUZIONI
5990 :
6000 POKEV,23:POKEV+1,49:POKEV+2
      1,1

```

```

6010 GOSUB7000:GOSUB7000
6020 FORW=1TO23:READU$:PRINTU$:G
      OSUB7000:NEXTW
6030 POKEV+21,0:POKEV,23:POKEV+1
      ,49:POKEV+23,0:POKEV+29,0
6040 GETA$:IFA$(<)CHR$(32) THEN604
      0
6050 RETURN
6060 :
6980 REM SOTTOPROGRAMMA--MOVIMEN
      TO SPRITE DURANTE LE
      ISTRUZIONI
6990 :
7000 FORI=1TO8:POKEV,PEEK(V)+1
7010 POKEV+1,PEEK(V+1)+1:NEXT
7020 POKEV,PEEK(V)+1:RETURN
7030 :
7980 REM SOTTOPROGRAMMA--ATTESA
7990 :
8000 POKESD+24,0
8010 SYSCH:IF PEEK(AR)<>32 THEN80
      10
8020 POKESD+24,15
8030 RETURN

```



joystick

Space Patrol

La tua missione è di annientare i dischi volanti provenienti dal pianeta Vencenlux, prima che



riescano ad atterrare in più di cinque sulla terra.

Con il tuo Joystick puoi muoverti in ogni direzione e sparare.

Stai molto attento perché la tua riserva di proiettili, che viene visualizzata sulla destra dello schermo, è limitata.

Quando i proiettili saranno terminati, dovrai atterrare su una base per ottenere una nuova scorta di munizioni.

Ogni 500 punti ti verrà regalata una vita. È possibile scegliere tra cinque difficoltà, ognuna delle quali prevede velocità e punteggio diverso per ogni disco volante colpito.

N.B. Il nome del secondo programma deve essere "SP".

**Aspettiamo
i tuoi lavori
migliori**

programma I

```
5 PRINT"{CLR}{ 7 GIU' }
  { 2 SPAZI}{RVS}** SPACE PATROL
  **{ 8 GIU' }"
11 PRINT" ATTENDI UN ATTIMO..."
15 POKE52,27:POKE56,27
20 FORI=7168TO7679:POKEI,PEEK(I+
  25600):NEXT
30 READX:IFX<0THEN45
35 FORI=XTOX+7:READJ:POKEI,J:NEX
  T:GOTO30
45 S$="LO"+CHR$(34)+"SP"+CHR$(34
  )+",1:"+CHR$(131)
50 FORI=1TOLEN(S$):POKE630+I,ASC
  (MID$(S$,I)):NEXT:POKE198,I:EN
  D
800 PRINT"{HOME}":NEW:CLR:END
1000 DATA7384,0,0,0,63,95,255,0,
  0
1001 DATA7392,2,6,14,254,254,255
  ,240,60
1002 DATA7400,64,96,112,127,127,
  255,15,60
1003 DATA7408,0,0,0,252,250,255,
  0,0
1004 DATA7416,0,0,60,126,171,126
  ,60,0
1005 DATA7432,0,16,84,16,254,16,
  84,16
1006 DATA7440,0,146,16,56,254,56
  ,16,146
1007 DATA7448,64,96,112,95,64,25
  5,19,62
1008 DATA7456,0,0,0,248,12,255,0
  ,0
1009 DATA7464,0,0,0,31,48,255,0,
  0
1010 DATA7472,2,6,14,250,2,255,2
  00,124
1011 DATA7480,0,0,0,0,0,126,0,0
1012 DATA7488,127,204,200,126,6,
  14,28,255
1013 DATA7496,254,51,19,126,96,1
  12,56,255
1014 DATA7504,0,0,2,3,15,31,63,2
  55
1015 DATA7512,1,3,7,15,159,255,2
  55,255
1016 DATA7520,0,128,192,192,224,
  227,247,255
1017 DATA7528,0,4,14,63,255,255,
  255,255
1018 DATA7536,4,6,15,191,255,255
  ,255,255
1019 DATA7544,0,48,242,255,255,2
  55,255,255
1020 DATA7632,0,0,0,129,195,231,
  255,255
1021 DATA7640,32,112,248,252,254
  ,255,255,255
1022 DATA7648,0,0,0,32,112,248,2
  52,255
1023 DATA7656,255,255,255,255,25
  5,255,255,255
1024 DATA7664,0,4,14,14,14,4,14,
  10
1025 DATA7672,0,0,0,0,0,0,0,0
1026 DATA6913,166,47,134,95,166,
  48,134,96
1027 DATA6921,160,0,177,95,201,2
  04,240,12
1028 DATA6929,24,101,95,133,95,1
  44,2,230
1029 DATA6937,96,76,9,27,96,0,0,
  0
1030 DATA6945,0,0,0,0,32,148,224
  ,165
1031 DATA6953,142,56,229,11,176,
  252,101,11
1032 DATA6961,133,11,96,0,24,32,
  148,209
1033 DATA6969,160,0,177,88,208,5
  8,162,22
1034 DATA6977,134,93,162,30,134,
  94,166,142
1035 DATA6985,16,16,162,9,134,11
  ,32,37
1036 DATA6993,27,169,12,24,101,1
  1,133,11
1037 DATA7001,144,7,162,10,134,1
  1,32,37
1038 DATA7009,27,165,93,24,101,1
  1,133,93
1039 DATA7017,160,1,145,88,144,2
  ,230,94
1040 DATA7025,165,94,136,145,88,
  76,163,27
1041 DATA7033,177,88,133,94,200,
  177,88,133
1042 DATA7041,93,160,22,177,93,2
  01,32,208
1043 DATA7049,41,169,32,160,0,14
  5,93,165
1044 DATA7057,93,24,105,22,133,9
  3,200,145
1045 DATA7065,88,136,144,6,230,9
  4,165,94
1046 DATA7073,145,88,169,31,145,
  93,165,94
1047 DATA7081,24,105,120,133,94,
  169,7,145
1048 DATA7089,93,96,48,253,169,3
  2,160,0
1049 DATA7097,145,93,169,0,145,8
  8,200,145
```

```

1050 DATA7105,88,141,0,27,76,178
,27,0
1051 DATA7113,0,32,1,27,160,1,17
7,95
1052 DATA7121,201,128,240,6,32,9
,27,76
1053 DATA7129,205,27,173,0,27,13
3,11,32
1054 DATA7137,37,27,230,11,165,1

```

```

1,141,0
1055 DATA7145,27,32,53,27,173,0,
27,240
1056 DATA7153,7,133,11,230,11,32
,53,27
1057 DATA7161,96,0,0,0,0,0,255,2
55
1058 DATA-1

```

programma 2

```

30 C=30720:V=36878:S1=36875:S2=S
1+1:S3=S1+2:HS=7664:HI=PEEK(H
S)*256+PEEK(HS+1)
40 DD=37154:DIML%(13):POKE36879,
9:GOSUB700:POKE36869,255
50 PRINT"{CLR}{WHT}{RVS}SC=0
{ 6 SPAZI}HI=0{ 5 SPAZI}#
{HOME}"
55 P=7976:D=0:DI=1:BA=20:OP=5:SC
=0:BO=0:G=7723:C$="{HOME}
{ 21 GIU'}"
60 FORI=38410TO38850STEP22:POKEI
,1:POKEI+1,1:NEXT:FORI=38443T
O38861STEP22:POKEI,5:NEXT
65 GOSUB380:GOSUB850:PRINTC$;A$;
"{CYN}@@@@@@@@@@@@@@@@@@@@@
{HOME}":POKEP,27:POKEP+1,28
70 POKE38905,3:POKE8185,0:GOSUB8
50:FORT=1TO6:GOSUB900:NEXT
75 FORH=1TO5-U
76 POKE7420,204:POKE37139,0:POKE
DD,127:J=PEEK(37152)AND128:J3
=-(J=.):POKEDD,255
90 J=PEEK(37137):J1=-((JAND8)=.)
:J2=-((JAND16)=.):J0=-((JAND4
)=.):FR=-((JAND32)=.)
95 IFFRANDDI=1THEN400
99 POKE7420,179
100 IFJ1THEN265
105 IFJ2THEN220
110 IFJ3THEN200
115 IFJ0THEN255
120 NEXT:GOSUB900:GOTO75
200 D=1:POKEP,32:POKEP+1,32:GOSU
B300:A$=MID$(A$,2,26)+MID$(A
$,1,1):GOTO250
220 D=0:POKEP,32:POKEP+1,32:GOSU
B300:A$=MID$(A$,27,1)+MID$(A
$,1,26)
250 PRINTC$;A$"{HOME}":GOTO120
255 POKEP,32:POKEP+1,32:P=P-22:I
FP<7712THENP=P+22
260 GOSUB300:GOTO120
265 POKEP,32:POKEP+1,32:P=P+22:I
FP>8130THENP=P-22
275 GOSUB300

```

```

280 IF(P=8130)ANDPEEK(P+22)=40AN
DPEEK(P+23)=41ANDDI=0THENGOS
UB350:DI=1:GOSUB300
285 GOTO250
300 IFD=.THEN315
305 IFDI=.THENPOKEP,35:POKEP+1,3
6:RETURN
310 POKEP,29:POKEP+1,30:RETURN
315 IFDI=.THENPOKEP,37:POKEP+1,3
8:RETURN
320 POKEP,27:POKEP+1,28:RETURN
350 POKEV,15:GOSUB855:FORT=150TO
250:POKES1,T:POKES2,T:NEXT
370 BA=BA-6+U:IFBA<5THENBA=5
380 B=BA:G=8163-BA*22:FORT=GTO81
41STEP22:POKET,61:NEXT
385 POKEV,0:POKES1,0:POKES2,0:RE
TURN
400 POKEV,15:POKES2,245:IFD=.THE
N430
410 F=P+2
415 IFF=P+11THEN493
420 IFPEEK(F)=31THEN460
425 POKEF+C,1:POKEF,39:FORT=1TO1
:NEXT:POKEF,32:F=F+1:GOTO415
430 F=P-1
435 IFF=P-11THEN493
440 IFPEEK(F)=31THEN460
445 POKEF+C,1:POKEF,39:FORT=1TO1
:NEXT:POKEF,32:F=F-1:GOTO435
460 X=0:FORT=0TO13:IFL%(T)=FTHEN
X=T:L%(T)=0
470 NEXT:POKES3,200:FORT=15TOOST
EP-5:POKEF+C,7:POKEF,33:POKE
V,T:POKEF+C,2:POKEF,34:NEXT
480 POKES3,0:POKEF,32:SC=SC+2+U*
4:BO=BO+2+U*4
485 IFBO>500THEN:GOSUB500:OP=OP+
1:BO=BO-500:GOSUB855
490 PRINT"{HOME}{WHT}{RVS}"TAB(3
)SC
493 B=B-1:IFB=0THENDI=0:GOSUB300

```



```

498 POKEG,32:G=G+22:POKEV,0:POKE
S2,0:GOTO120
500 POKEV,15:FORT=1TO28:PRINT"
{HOME}{GIU'}{WHT}";MID$("
{ 18 SPAZI}{E''BONUS ",T,22)
:POKES2,230
510 FORY=1TO85:NEXTY,T:POKEV,0:P
OKES2,0:RETURN
700 POKE36869,240:PRINT"{CLR}
{ 5 GIU'}{ 2 DES}{RVS}{YEL}*
* SPACE PATROL **"
710 PRINT"{GRN}{ 6 GIU'}{DES}FAC
ILE{ 5 DES}DIFFICILE "
715 PRINT"{GIU'}{ 4 DES}{RVS}
{WHT}1{OFF}..{RVS}2{OFF}..
{RVS}3{OFF}..{RVS}4{OFF}..
{RVS}5{OFF}"
720 GETW$:IFW$=""THEN720
725 U=VAL(W$):IFU<1ORU>5THEN720
730 Q%=14-2*U:RETURN
850 PRINT"{HOME}{WHT}{RVS}"TAB(1
3)HI
855 A$="{CYN}/:/;<{WHT}() {CYN}*+,
-./:/;<{WHT}() {CYN}*+,-."
860 PRINT"{HOME}{WHT}{RVS}"TAB(2
0)OP:RETURN

```

```

900 POKE6912,Q%:SYS7114:IFPEEK(6
912)=0THEN950
910 RETURN
950 OP=OP-1:GOSUB860
960 POKEV,15:FORT=1TO15:POKES2,2
50:FORY=1TO5:NEXT:POKES2,0:N
EXT:POKEV,0
985 IFOP<>0THENRETURN
990 POKEV,15:FORT=8TO15:POKE3687
9,25:FORY=200TO220+T*2:POKES
2,Y:NEXT:POKE36879,9:NEXT
1000 POKES0,0:POKES2,0:PRINT"
{HOME}{ 5 GIU'}"TAB(4)"
{WHT}++++++"
1005 PRINTTAB(4)"← GAME OVER ←":
PRINTTAB(4)"++++++"
1006 PRINT"{ 3 GIU'}"TAB(4)"
{WHT}PREMI UN TASTO"
1010 IFSC>HITHENHI=SC:GOSUB850
1020 POKE7420,204:FORY=1TO80:NEX
T:POKE7420,51:FORY=1TO80:NE
XT
1025 GETW$:IFW$=""THEN1020
1040 POKEHS,INT(HI/256):POKEHS+1
,(HI/256-INT(HI/256))*256
1050 RUN

```

VIC Organetto



Dopo un'esplosiva introduzione, potrai suonare anche tu con un magnifico organetto.

Durante l'esecuzione di qualsiasi motivo musicale, potrai usufruire di varie opzioni:

- premendo "F1" "F3" "F5" "F7" puoi selezionare la voce con cui suonare;

- premendo "F2" "F4" "F6" "F8" puoi sovrapporre più di una voce.

Per finire premi '←'.

N.B. È 'indispensabile' mettere nel nome del

secondo programma (prima delle parole) un simbolo di colore che non sia "CYN" e "WHT".

Per ciò il comando "SAVE" dovrà avere la seguente forma:

- "SAVE" (comando colore) (nome programma).

Il nome del secondo programma deve essere "V.O."

Buon divertimento!!!!

programma 1

```

10 PRINT"{CLR}":A=36874:B=36875:
C=36876:D=36877:E=36878:Q=15:
W=0
12 PRINT"{ 9 DES}{RVS} {DES}
{OFF}"
13 PRINT"{ 9 DES}{RVS} {DES}
{OFF}"

```

```

14 PRINT"{ 9 DES}{RVS} {DES}
{OFF}"
15 PRINT"{ 9 DES}{RVS} {DES}
{ 2 SPAZI}{OFF}"
20 PRINT"{HOME}{ 7 GIU'}{RED}
{ 4 DES}{RVS} { 3 DES}
{ 2 DES} { 2 DES}{ 3 SPAZI}

```

```

{OFF}"
21 PRINT"{ 4 DES}{RED}{RVS}
{ 3 DES} { 2 DES} { 2 DES}
{OFF}"
22 PRINT"{ 4 DES}{RED}{RVS}
[<*>]{DES}E { 2 DES}
{ 2 DES} {OFF}"
23 PRINT"{ 4 DES}{RED}[<*>]{RVS}
{ 3 SPAZI}{OFF}E{RVS}
{ 2 DES} { 2 DES} {OFF}"
24 PRINT"{ 5 DES}{RED}[<*>]{RVS}
{OFF}E{RVS}{ 3 DES}
{ 2 DES}{ 3 SPAZI}{OFF}"
25 PRINT"{GIU'}{GRN}{ 10 DES}&":
PRINT"{ 2 GIU'}{BLK}{ 7 DES}
ANTICOLI":PRINT"{GIU'}
{ 5 DES}MASSIMILIANO"
26 PRINT"{PUR}{ 2 GIU'}{ 5 DES}
PRESENTANO{DES}:"
30 FORT=1TO255:POKE36879,T:NEXTT
31 FORM=1TO2000:NEXT:POKE36879,5
9
32 PRINT"{CLR}{ 4 GIU'}{RED}
{ 4 DES}{RVS} { 3 DES}
{ 2 DES} { 2 DES}{ 3 SPAZI}
{OFF}"
33 PRINT"{ 4 DES}{RED}{RVS}
{ 3 DES} { 2 DES} { 2 DES}
{OFF}"
34 PRINT"{ 4 DES}{RED}{RVS}
[<*>]{DES}E { 2 DES}
{ 2 DES} {OFF}"
35 PRINT"{ 4 DES}{RED}[<*>]{RVS}
{ 3 SPAZI}{OFF}E{RVS}
{ 2 DES} { 2 DES} {OFF}"
36 PRINT"{ 5 DES}{RED}[<*>]{RVS}
{OFF}E{RVS}{ 3 DES}
{ 2 DES}{ 3 SPAZI}{OFF}"
50 PRINT"{ 5 GIU'} [A][S]
[A][S][A][S][A]
[S][M]M[G][A]* PO PO
[A][S]"
60 PRINT" { 2 -}[Q][X]-[S]
[Q][W][M] [G][Q]
{ 2 SPAZI}[M][G] [M]
[G]{ 2 -}";
70 PRINT" [Z][X]-M[Z][X]

```

```

{ 2 -}[M] [G][Z]* [M]
[G] [M][G][Z][X]";
305 POKEE,Q:FORL=1TO5:POKEB,200:
FORM=1TO500:NEXTM:POKEB,W
310 POKEC,200:FORM=1TO500:NEXTM:
POKEC,W:NEXTL:POKEE,W
320 POKEE,Q:FORL=1TO5:FORM=180TO
235STEP2:POKEC,M:FORN=1TO10
330 NEXTN,M:POKEC,W:FORM=1TO100:
NEXTM,L:POKEE,W
340 POKED,220:FORL=15TO0STEP-1:P
OKEE,L:FORM=1TO300:NEXTM,L
350 POKED,W:POKEE,W
360 POKEE,Q:FORL=1TO100:POKEC,IN
T(RND(1)*128)+128:FORM=1TO10
370 NEXTM,L:POKEC,W:POKEE,W
380 POKEE,10:FORL=230TO128STEP-1
:POKEC,L:FORM=1TO20:NEXTM,L
400 POKEC,W:POKED,200:FORL=15TO0
STEP-.05:POKEE,L:NEXTL:POKED
,W
410 POKEE,Q:FORL=1TO30:FORM=250T
O240STEP-1:POKEC,M:NEXTM
420 FORM=240TO250:POKEC,M:NEXTM:
POKEC,W:NEXTL:POKEE,W
430 POKEE,Q:FORL=130TO254:POKEC,
L:FORM=1TO40:NEXTM,L
440 POKEE,W:POKEC,W
450 POKEE,Q:FORL=1TO20:FORM=220-
LTO160-LSTEP-4:POKEC,M
460 NEXTM:FORM=160-LTO220-LSTEP4
:POKEC,M:NEXTM,L
470 POKEE,W:POKEC,W
480 POKEE,15:FORL=1TO15:FORM=200
TO220+L*2:POKE36876,M:NEXTM,
L
490 POKEE,W:POKEC,W
500 PRINT"{ 2 GIU'}{RED}
{ 4 SPAZI}PREMI UN TASTO"
600 GETAS:IFAS=""THEN600
650 PRINT"{CLR}"
700 S$="LO"+CHR$(34)+"V.O."+CHR$(
34)+"",1:"+CHR$(131)
750 FORI=1TOLEN(S$):POKE630+I,AS
C(MID$(S$,I)):NEXT:POKE198,I
:END

```

programma 2

```

1 GOSUB10000
10 POKE36878,F:K=0
11 IFF=0THEN15
12 F=F-1.5
15 POKES,R
20 GETAS
25 IFAS="Q"THENR=135:K=1

```

```

30 IFAS="2"THENR=143:K=1
35 IFAS="W"THENR=147:K=1
40 IFAS="3"THENR=151:K=1
45 IFAS="E"THENR=159:K=1
46 IFAS="R"THENR=163:K=1
47 IFAS="5"THENR=167:K=1
48 IFAS="T"THENR=175:K=1

```

```

49 IFA$="6"THENR=179:K=1
50 IFA$="Y"THENR=183:K=1
51 IFA$="7"THENR=187:K=1
52 IFA$="U"THENR=191:K=1
53 IFA$="I"THENR=195:K=1
54 IFA$="Z"THENR=195:K=1
55 IFA$="S"THENR=199:K=1
56 IFA$="X"THENR=201:K=1
57 IFA$="D"THENR=203:K=1
58 IFA$="C"THENR=207:K=1
59 IFA$="V"THENR=209:K=1
60 IFA$="G"THENR=212:K=1
61 IFA$="B"THENR=215:K=1
62 IFA$="H"THENR=217:K=1
63 IFA$="N"THENR=219:K=1
64 IFA$="J"THENR=221:K=1
65 IFA$="M"THENR=223:K=1
66 IFA$=","THENR=225:K=1
79 IFA$="+"THENPOKE36878,0:END
90 IFA$="{F1}"THENK=1:GOSUB8000
91 IFA$="{F2}"THENK=1:GOSUB6900
92 IFA$="{F4}"THENK=1:GOSUB7000
93 IFA$="{F6}"THENK=1:GOSUB7100
94 IFA$="{F8}"THENK=1:GOSUB7200
95 IFA$="{F3}"THENK=1:GOSUB8100
96 IFA$="{F5}"THENK=1:GOSUB8200
99 IFA$="{F7}"THENK=1:GOSUB8300
100 IFA$<>"ANDK=1THENF=15
200 GOTO10
6900 S=36874:RETURN
7000 S=36875:RETURN
7100 S=36876:RETURN
7200 S=36877:RETURN

8000 POKES,0:FORR=1TO100:S=36874
:RETURN

8100 POKES,0:FORR=1TO100:S=36875
:RETURN

8200 POKES,0:FORR=1TO100:S=36876
:RETURN

8300 POKES,0:FORR=1TO100:S=36877
:RETURN

10000 PRINT"{CLR}";:POKE36879,26

10010 FORZ=1TO44:PRINT"{BLK}
{RVS}{ 4 SPAZI}";:NEXT
10020 PRINT"{HOME}{ 4 GIU' }
{ 4 DES}{RVS}{BLK}VIC. ORG
ANETTO";
10040 PRINT"{HOME}{ 8 GIU' }{BLU}
[< 22 X>]";
10050 PRINT"{BLK}{OFF}
{ 2 SPAZI}{RVS}{ 3 SPAZI}

```

```

[<G>]{ 2 SPAZI}{OFF}
{ 3 SPAZI}{RVS}{ 3 SPAZI}
[<G>]{ 2 SPAZI}[<G>]
{ 2 SPAZI}{OFF}{RVS}
{OFF}{ 2 SPAZI}{RVS} 2
[<G>]3 {OFF}{ 3 SPAZI}
{RVS} 5 [<G>]6 [<G>]7
{OFF}{RVS} ";
10060 PRINT"{OFF}[< 2 T>]P
[< 2 T>]P[< 2 T>]P
[< 2 T>]P[< 2 T>]P
[< 2 T>]P[< 3 T>]{RVS}
{OFF}{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}{RVS}
{OFF} Q[<M>] W[<M>] E[<M>]
R[<M>] T[<M>] Y[<M>] U
{RVS} ";
10070 PRINT"{OFF}{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 3 SPAZI}{RVS} ";
10080 PRINT"{BLU}{OFF}[< 22 X>]
";
10090 PRINT"{BLK}{OFF}
{ 2 SPAZI}{RVS}{ 3 SPAZI}
[<G>]{ 2 SPAZI}{OFF}
{ 3 SPAZI}{RVS}{ 3 SPAZI}
[<G>]{ 2 SPAZI}[<G>]
{ 2 SPAZI}{OFF}{RVS}
{OFF}{ 2 SPAZI}{RVS} S
[<G>]D {OFF}{ 3 SPAZI}
{RVS} G [<G>]H [<G>]J
{OFF}{RVS} ";
10100 PRINT"{OFF}[< 2 T>]P
[< 2 T>]P[< 2 T>]P
[< 2 T>]P[< 2 T>]P
[< 2 T>]P[< 3 T>]{RVS}
{OFF}{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}{RVS}
{OFF} Z[<M>] X[<M>] C[<M>]
V[<M>] B[<M>] N[<M>] M
{RVS} ";
10110 PRINT"{OFF}{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 2 SPAZI}[<M>]
{ 2 SPAZI}[<M>]{ 2 SPAZI}
[<M>]{ 3 SPAZI}{RVS} ";
10120 PRINT"{BLU}{OFF}[< 21 X>]
";
10130 POKE8185,102:POKE38905,6
30000 RETURN

```

Bee Trap

Un prato è stato invaso da api venusiane che vogliono prendere il predominio della zona dove si trovano.

Esse sono molto attraenti e le api terrestri, vedendole, si dirigono subito verso di esse e verso la morte sicura.

Fortunatamente l'ape regina vi ha inviato a salvare le api vaganti portandole nell'aiuola, nella parte alta dello schermo.

Facendo ciò guadagnerete un punto per ogni ape salvata, ma ne perderete uno per ogni ape

annientata dagli ospiti extraterrestri (le api venusiane).

Se poi anche voi, vagando per il prato, andrete a finire su un'ape venusiana, morirete e la partita avrà termine.

Per muovere l'ape dovrete usare il Joystick e dovrete stare molto attenti perché il numero di api venusiane presenti sul campo aumenta continuamente.

Buona fortuna e... attenti alle api extraterrestri!

```

3 GOTO10
5 J1=NOTPEEK(37151)AND60-((PEEK(
  JS)AND128)=0)
7 RETURN
10 Q=36879:J=37137:JS=37152
30 POKEQ,110:PRINT"{CLR}
  { 6 GIU'}{WHT}INSERISCI IL JO
  YSTICK!{ 2 GIU'}PREMI IL TAST
  O 'FIRE'."
32 POKE37154,127
35 GOSUB5
40 IFJ1<>32THEN35
50 PRINT"{CLR}":POKEQ,93:FORL=1T
  O3:PRINT"{ 3 GIU'}{ 5 DES}B E
  E{ 3 SPAZI}T R A P{ 2 GIU'}
  ";NEXT:FORL=1TO999:NEXT
60 POKEQ,127:FORL=1TO999:NEXT:PO
  KEQ,42:FORL=1TO2000:NEXT
80 PRINT"{CLR}{BLK} TU SEI UN' A
  PE-GUIDA{ 2 SPAZI}E L'APE REG
  INA TI HA{ 2 SPAZI}INVIATA IN
  { 2 SPAZI}UN PRATO"
90 PRINT" A SALVARE LE API CHE
  { 5 SPAZI}SI SONO PERSE,"
95 PRINT"PORTANDOLE NELL'AIUOLA-
  ----- OGNI AP
  E{ 2 SPAZI}SALVATA TI"
100 PRINT"FRUTTA 1 PUNTO, MENTRE
  { 2 SPAZI}OGNI APE CHE VIENE
  { 2 SPAZI}INTRAPPOLATA DA UN
  'APE"
105 PRINT"{SU}VENUSIANA TI TOGLI
  E UN{ 8 SPAZI}PUNTO."
110 PRINT"{ 2 SPAZI}STAI ATTENTO
  A NON{ 4 SPAZI}FARTI INTRAP
  POLARE{ 3 SPAZI}DALLE
  { 2 SPAZI}ATTRAENTI API"

```

```

111 PRINT"{ 2 SPAZI}VENUSIANE CH
  E SONO{ 4 SPAZI}PRESENTI IN
  GRANDE{ 3 SPAZI}QUANTITA' SU
  L CAMPO."
112 PRINT"{ 4 SPAZI}BUONA FORTUN
  A!{GIU'}"
115 FORL=1TO200:NEXT
120 PRINT"{WHT}{ 4 SPAZI}{RVS}(P
  REMI 'FIRE')"
130 GOSUB5:IFJ1<>32THEN130
150 C=30720:P=7680:N=36877:V=368
  78:B=36875:K=8248:HH=36874
175 K1=7735:H=240:SC=0:S=0:D1=0:
  D2=0:R=0
200 POKEQ,191:PRINT"{CLR}":FORL=
  0TO43
225 POKEP+L,42
230 POKEP+L+C,INT(RND(1)*8):NEXT
305 POKEK1+D1,32
312 J1=NOTPEEK(37151)AND60-((PEE
  K(JS)AND128)=0)
315 IFPEEK(JS)=119THEND1=D1+1:GO
  TO335
320 IFJ1=4THEND1=D1-22:GOTO335
325 IFJ1=8THEND1=D1+22:GOTO335
330 IFJ1=16THEND1=D1-1
335 IFK1+D1>8185THEND1=450
340 IFK1+D1<7724ANDS>0THEN2000
345 IFK1+D1<7724THEND1=0:POKEQ,1
  10:FORL=1TO2000:NEXT:SC=SC-1
  :POKEQ,191
350 IFD2<50ANDX<2THENXX=INT(RND(
  1)*351)+44:POKEK1+C+XX,X+12:
  POKEK1+XX,88
360 IFPEEK(K1+D1)=42THEND2=0:S=S
  +1:H=H-10

```

```

370 IFPEEK(K1+D1)=88THEN3000
380 POKEK1+D1+C,8:POKEK1+D1,42
440 IFH<130THENH=240
450 POKEV,6-X/6:POKEB,160-X:POKE
HH,H
505 POKEK-D2,32
520 X=INT(RND(1)*(H/5-22))
530 IFX<3THEND2=D2+22
540 IFX=3ORX=4THEND2=D2+1
550 IFX=5THEND2=D2-1
560 IFK-D2<7724THEND2=0
570 IFPEEK(K-D2)=42ANDD2>62THEND
2=0:S=S+1:H=H-10
580 IFPEEK(K-D2)=88THEN1000
590 POKEK-D2+C,14:POKEK-D2,42
600 GOTO305

1000 POKEQ,42:FORL=1TO10STEP.1:P
OKEV,10-L:POKEN,200-L*5:NEX
T:POKEN,0:D2=0
1050 R=R+1:IFR=50THEN3000
1075 IFR>40THENPOKEQ,249:GOTO305

1100 POKEQ,191:GOTO305
2000 POKEQ,127:SC=SC+1:PRINT"
{HOME}{ 2 GIU'}{ 6 DES}
{RED}HAI"SC;

```

```

2100 IFSC=1THENPRINT" PUNTO":GOT
O2300
2200 PRINT" PUNTI":GOTO2300
2300 FORL=1TO75STEP.3:POKEB,175+
L:NEXT:PRINT"{HOME}
{ 2 GIU'}{ 22 SPAZI}":S=0:D
1=0
2310 GOTO1100
3000 POKEQ,10:FORL=1TO10STEP.1:P
OKEV,15-L:POKEB,200-L*5:NEX
T:POKEB,0:POKEHH,0
3100 PRINT"{HOME}{ 3 GIU'}IL GIO
CO E' TERMINATO!{GIU'}
{ 2 SPAZI}HAI SALVATO"SC;
3200 IFSC=1THENPRINT" APE":GOTO3
400
3300 PRINT" API"
3400 PRINT"{ 2 GIU'} PREMI {RVS}
F{OFF} PER FINIRE O {GIU'}
{ 2 SPAZI}'FIRE' PER GIOCAR
E"
3403 GETA$
3410 IFA$="F"THEN3500
3415 GOSUB5:IFJ1=32THEN150
3420 GOTO3403
3500 POKE37154,255:POKEQ,27:PRIN
T"{CLR}{BLU}{GIU'}":END

```

Per ricevere gli arretrati di **PAPER soft** compilare il seguente tagliando:

J.soft **Compilare ed inviare in busta chiusa a:**
via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di

PAPER soft _____

a L. 2.000 cad. per un totale di L. a mezzo

assegno allegato


contanti allegati



il Corso di BASIC in meno di 6 mesi ideato dagli specialisti del Gruppo Editoriale Jackson



ché il calcolatore da solo non sa fare quasi niente. I calcolatori non sono costruiti per svolgere subito un calcolo come avviene per le calcolatrici; sono invece predisposti per eseguire un programma che dobbiamo preparare in precedenza e che descrive in dettaglio tutto ciò che il calcolatore dovrà fare. Una volta che questo programma è pronto, allora può essere eseguito e anche memorizzato per un suo eventuale e ripetuto uso successivo. Abbiamo definito i calcolatori come macchine da calcolo automatiche a programma memorizzato. In questa definizione è racchiuso tutto ciò che si può dire sui calcolatori: che sono macchine automatiche come tante altre macchine, che sono da calcolo come le calcolatrici, ma soprattutto che lavorano con un programma che deve essere in precedenza memorizzato. Un programma adeguato "trasforma" il calcolatore nella macchina specializzata in cui abbiamo bisogno in quel momento. Senza programma, il calcolatore non può fare nulla, come un'auto senza conducente. In Figura 4 mostriamo alcuni esempi di programmi scritti con un diverso linguaggio comprensibile dal calcolatore. Torniamo ora alla definizione di calcolatore che avevamo dato all'inizio: il calcolatore che ha la nostra intelligenza e si opera con l'uomo in fatto di programmi. I nostri bracci, sono costruiti con circuiti integrati. In effetti, si potrebbe dire che i calcolatori sono molto stupide, in quanto nulla che non sia "programmato" in loro. Il primo "cervello" di un calcolatore è un "chip" di silicio. Per effettuare questa operazione si usa l'istruzione SAVE descritta in dettaglio nel relativo riquadro. Poiché l'uso di SAVE non è uguale in tutte le macchine, vi consigliamo di consultare il manuale del computer che state usando. Il trasferimento inverso del programma, dalla memoria esterna a quella centrale, si effettua con l'istruzione LOAD (in alcune versioni BASIC si usa l'istruzione OLD che è equivalente a LOAD).



ABC PERSONAL COMPUTER

Insegna tutto quel che c'è da sapere sul BASIC;
mette subito in grado di programmare;
fornisce una guida ragionata e competente alla scelta del Personal Computer.

ABC Personal Computer
è il corso che risponde oltretutto alle esigenze di chi non ha tempo da perdere, perché:
si completa in meno di 6 mesi
si compone di 24 fascicoli settimanali

ABC

per tutti coloro che vogliono davvero "dialogare" col computer

FASCICOLI SETTIMANALI
24
IN EDICOLA
DAL 21-9-84

La rapida diffusione dei computer nelle aziende, nelle scuole, nelle case ha reso urgente e indispensabile per molti imparare il linguaggio del calcolatore. Ci sono molti modi per farlo. Il più nuovo, originale, rapido e divertente si chiama...

ABC Personal Computer

l'opera creata per rispondere alle esigenze di chi, per lavoro, studio, hobby vuole acquisire un'effettiva padronanza dei piccoli elaboratori per sfruttarne le immense, entusiasmanti possibilità.

ABC Personal Computer

svela i segreti del BASIC - il linguaggio fondamentale dei personal, home e microcomputer - e fornisce la chiave per programmare da soli.

ABC Personal Computer

tratta il BASIC in modo diverso, organico, comprensibile a tutti perchè nasce dalla grande esperienza e dalla capacità divulgativa del Gruppo Editoriale Jackson.

ABC Personal Computer

è un'opera preziosa, comprensibile per i ragazzi, perchè comincia proprio dall'abc dell'informatica; ideale per i professionisti, perchè rapida, esauriente, concreta; preziosa per gli appassionati perchè ricca di programmi già testati, di suggerimenti, di notizie sul mondo dei piccoli computer.



GRUPPO EDITORIALE JACKSON
noi l'informatica
la conosciamo davvero

Personal Computer

24 appuntamenti in edicola per arricchire la biblioteca di casa con:

- 2 volumi di Lezioni per complessive 608 pagine
- 1 volume di Computer-test
- 1 Dizionario di Informatica di oltre 208 pagine

L'opera è elegantemente rilegata in similpelle.

Centinaia di illustrazioni a colori, foto e disegni.

Formato dei volumi cm. 21x28.

Formato del Dizionario di Informatica cm. 15x21.

Abbonamento-risparmio

Tagliando da inviare in busta chiusa a:
Gruppo Editoriale Jackson "ABC Personal Computer"
via Rosellini, 12 - 20124 Milano

Si, desidero sottoscrivere l'abbonamento risparmio ai 24 fascicoli di ABC Personal Computer e alle copertine dei 4 volumi dell'opera. Tutto al prezzo speciale di L. 80.000 invece di L. 96.000.

Allego alla presente

- assegno non trasferibile di L. 80.000 a voi intestato
- fotocopia di versamento di L. 80.000 sul ccp n. 11666203
- fotocopia di vaglia postale di L. 80.000 a voi intestato

I fascicoli dovranno essere inviati a

Nome _____ Cognome _____

Via _____

Città _____ Prov _____ C.A.P. _____

Data _____ Firma _____



IN MILANO - VIA MASCHERONI, 14

IL VOSTRO "NUOVISSIMO" COMPUTER SHOP

Libri e riviste di elettronica e informatica.

CORSI CONTINUI
TUTTO L'ANNO
CON I MIGLIORI
SOFTERISTI



IL TELEFONO È 02-437.385